Maple6の基本操作と実践 第4版

西谷滋人
京都大学工学研究科材料工学科
(version 4 for Release 6)

Department of Materials Science and Engineering,
Kyoto University, Kyoto 606-8501, Japan
e-mail : bob@karma.mtl.kyoto-u.ac.jp

平成13年4月6日

本書は PowerBookG3 上で Maple 6 と L^AT_EX を使用して編集，整形を行ないました.

Maple6 の基本操作と実践 第４版

Maple 6: Essentials and Applications on Materials Science

まえがき

> ”Any sufficiently advanced technology is indistinguishable from magic.“

Arthure C. Clarke の言葉です．Mathematica という数式処理ソフトを初めて使ったときにもこれと同じ感覚を持ちました．数式処理には大型計算機センターの Reduce を使うしかなかった当時に，私の持っていた MacSE で複雑な数式を一瞬にして簡単化し，数式らしく表示してくれたのを目のあたりにしたとき，また一歩科学が魔法に近づいた気がしました．

本書で使用法を解説する Maple は 1985 年に，Mathematica に先駆けて市場に現れました．カナダの Waterloo 大学と ETH Zürich での研究成果をもとにした計算ライブラリの内部処理は公開されており，その数式処理に対する信頼性の高さで定評があります．日常，私が Maple をどのように使っているかというと，紙と鉛筆，電卓，グラフソフト，programming 言語の代わりとして,（1）論文を読むときの数式の導出　（2）ちょっとしたルーチン計算,（3）結果のグラフ化，そして（4）プログラムを作るときのプロトタイプの作成，等です．

Mandelbrot 集合 を描くという実例を取り上げてみましょう．専門書をひもとくと (「C 言語によるアルゴリズム事典」，奥村晴彦著，技術評論社 1991）

> Mandelbrot(マンデルブロート）集合　Mandelbrot set
> (中略)
> 計算機では，ある領域の点 (x,y) について，
> z <-- x + i y; count <-- M;
> while (|z| <= 4) and (count>0) do begin
> z <-- z2 - (x + i Y); count <-- count -1
> end;
> 点 (x,y) に count で決まる色 (count=0 なら黒) を付ける;
>
> とする．黒い部分が Mandelbrot 集合で，それ以外の色は，その点と Mandelbrot 集合との ”近さ ”を表す光背である．

となっています．この後，プログラムコードが続き，どこかにある Graphics 表示用の謎のルーチンを呼ぶように指示してあります．これを Maple で実現しようと思うと，

```
> mandel:=proc(x,y)
local z0,z,count;
z0:=x+y*I;
z:=0;
count:=20;
while (evalf(abs(z))<4.0 and count>0) do
z:=z^2-z0;
count:=count-1;
od;
count;
end:
> with(plots):
densityplot(mandel,-1..2.5,-1.5..1.5,axes=none,colorstyle=HUE,
style=patchnogrid,grid=[50,50]);
```



となります．どうです?!　数学的な記述そのままで，しかもわずか数分で目的とする Mandelbrot 集合を実際に描くという作業が終了してしまいます．

数式処理や関数のグラフ化等を扱う優秀な市販ソフトウェアとしては Mathematica と Maple が有名です．どちらかの使い方になじんでいればもう一方の，あるいは今後現れるであろうより優秀なソフトを修得するのにそれほど時間はいらないと思います．私自身も Mathematica から Maple へ乗り換えるのに 1ヶ月程ですみました．本書があれば 1 週間で乗り換えられたと確信しています．

本書ではソフトの詳しい構造や数学の厳密な適用などには余り注意を払っていません．とにかく問題を理解し，解いていくために Maple をいかに使用するか，つまり道具として数学 (Maple) をいかに使うかに力点を置いて解説しています．ちゃんとした文法やコマンド引数などは help を参照し，試行錯誤をお願いします．

Maple で作った書類は L^AT_EX や HTML 等にも簡単に変換してくれます．本書は実際に L^AT_EX で出力したファイルから加工しています．また，本書の全てのスクリプトを HTML で

`http://www.mtl.kyoto-u.ac.jp/Maple/Maple.index`

から引けるようになっています．

本書は Maple6 に基づいて解説しています．MapleV からの変更点は

- Excel との共同作業の強化 (Windows 版のみ)．
- NAG ライブラリーによる行列の数値計算の高速化．
- プログラミング言語の変更．
- グラフ出力の強化．

です．上位互換ですので，古い教科書 (つまり知識) がそのまま使えます．古いリリースをお使いのときに，例外的に影響があるのは，readlib という関数を呼び出すコマンドがあったということです．また MapleV のリリース 4 からリリース 5 への変更で影響が大きい点は二ヶ所，

- 直前の出力を参照する記号が”から%に変わった．
- 文字列を囲む記号がバッククォート (‘) からダブルクォート (”) になった．

です．その他の変更点はヘルプの ”What’s new ”から引けます．

優れた題材とテキストを提供下さった材料工学科の沼倉宏先生と河合潤先生に感謝いたします．特に 1.6 節の題材は河合潤先生の手によるものです．また，Maple の導入・管理では応用システム科学教室の河野浩之先生のお世話になりました．心より感謝いたします．このテキストは河野先生の強力な後押しで完成することができました．なお，誤りや不正確な記述が多々あると思いますが，適宜訂正していきますので，e-mail で御連絡下さい．

目 次

第1章　基本操作

1.1　最初の一歩

ここでは Maple の基礎的な操作法と注意事項を述べます.

1.1.1　Maple の起動法

PC ではほかのソフトウェアと同様の立ち上げ方で起動します．unix では xmaple とすると GUI 版の maple が立ち上がります．maple だけですと普通の vt100 上での text 版が立ち上がります．これを unix の redirection 機能を使って filter として機能させることも可能です (付録 B.1参照)．起動できない場合は PATH と executability を確認して下さい.

1.1.2　簡単な演算

それでは簡単な計算を実行させてみましょう.

```
> 1+1;
```

$$2$$

enter と shift+enter は違った意味を持ちます．enter は入力，shift+enter は改行です．複数行にまたがる入力では改行には shift+enter を入力し，最後に enter をいれればその領域すべてを一度に入力したことになります.

```
> set1:={1,2,3};(shift+enter)
> set2:={3,4};(enter)
```

$$set1 := \{1, 2, 3\}$$

$$set2 := \{4, 3\}$$

これ以降の記述では最後の (enter) や (shift+enter) を省きます．最後の；(セミコロン) を忘れがちです．出力させたくないときには最後の；を：(コロン) にすれば，なにも出力しません．ただし，内部での代入は実行されています．入力の順番はエンターをいれた順番であり，画面の上下とは関係ありません．また入力領域のどの位置にカーソルがあってもエンターでその領域全部を入力したことになります．次に関数のプロットをしてみましょう.

```
> plot({erf(x),diff(erf(x),x)},x=-5..5);
```



となります．意味は直観的で，“erf とその微分 (diff) を-5 から 5 まで表示する” です．

1.1.3 間違い修正

打ち間違いなどの訂正はアローキー，あるいはマウスのクリックによってできます．後は訂正して enter を入れれば入力されます．ある領域を選択して，削除・カットおよびペーストなどの作業もマウスを使ってできます．例えばサイン関数をーπから＋πまでプロットしようとした時

```
> plot({sin(x)},x=-pi..pi);
```

```
Error, (in plot/transform) cannot evaluate boolean: -pi < -4
```

という警告が出ました．これは Maple では大文字と小文字を区別しているためにおこったことです．そこで，pi→Pi と修正すると無事表示されます．

```
> plot({sin(x)},x=-Pi..Pi);
```



1.1.4 実行の中断

enter を押した後，入力ミスに気づいて計算をやめさせたいときには中断ができます．PC では tool bar の stop マークで，そして unix では interrupt button で中断します．

1.1.5 ヘルプファイル

```
> ?help;
```

でキーワードに関するヘルプが表示されます．その他，?? や？index あるいは？（キーワードの最初の一部）を使って，類推によってキーワードの情報を取り出すことができます．その他の help に関する操作はメニューバーの ”Help ”にいくつか用意されています．以下は help の出力です．記述は Windows 版を除いて英語です．英語が分からなくても Examples を参考にすればだいたい予測できます．

```
> ?plot;
```

plot - *create a two-dimensional plot of functions* （関数の説明）

Calling Sequence:（呼び出し）

plot(f, h, v)

plot(f, h, v,...)

Parameters:（引数の説明）

f - function(s) to be plotted

h - horizontal range

v - vertical range (optional)

Description:（詳しい解説）

- A typical call to the plot function is plot(f(x),x=a..b), where f is a real function in x and a..b specifies the horizontal real range on whi ch f is plotted.

⋮
中略
⋮

Examples:（使用例）

```
> plot(cos(x) + sin(x), x=0..Pi);
> plot(tan(x), x=-Pi..Pi);
> plot([sin(t), cos(t), t=-Pi..Pi]);
> plot(sin(t),t);
```

Since no domain is specified in the last example, we use t=-10..10.

⋮
中略
⋮

See Also:（関連する項目）

plot[spec] where spec is one of ***infinity***, ***polar***, ***parametric***, ***multiple***, ***ranges***, ***function***, ***options***, ***structure***, ***setup***, ***device***, ***replot***, ***style***, ***color***. Use ***plot3d*** for plotting surfaces. See ***discont*** and ***fdiscont*** regarding discontinuities.

1.2 簡単な数式処理とプロット

簡単な数値の代入と関数，関数の定義と式の変形，グラフのプロットを扱います．式の変形は難しく，なかなか思うように単純化してくれません．人間だとすぐに分かることも，Maple に教えてやらなければなりません．しかし，Maple はややこしい式の変形でも係数や次数を見落とすことはありません．実践演習 [2.2] を参考にうまいやり方を身に付けて，Maple を積極的に活用して下さい．

1.2.1 変数，式の定義とそのキャンセル

Maple の基本的な代入は

```
> mass:=10;
```

$$mass := 10$$

によって行われます．あらかじめ Maple で定義されていてよく使う定数は Pi, I, infinity などです．(?ininames 参照) 式の定義も同様に行えます．

```
> force:=-mass*accel;
```

$$force := -10\, accel$$

直前の結果を参照するには%を使います．

```
> exp1:=%;
```

$$\mathrm{exp1} := -10\, accel$$

これら一度数値を入れた変数を元へ戻すには

```
> restart;
```

によって行います．これで起動初期のなにも入力されていない状態に戻ります．ひとつの定義だけを初期状態に戻したいときには ’(シングルクォート）をもちいて

```
> mass:='mass';
```

$$mass := mass$$

によって行います．これによって，

```
> force:=-mass*accel;
```

$$force := -mass\, accel$$

となります．一時的な代入は subs で行います．

```
> subs(mass=10,accel=14,force);
```

$$-140$$

こうすると，

```
> force;
```

$$-mass\, accel$$

とそれぞれの変数が数値を取るのではなく，変数のままで扱われます．逆に一時的に値ではなく変数そのものを使用するにもシングルクォートを使います．

```
> x:=2;y:=3;
```

$$x := 2$$
$$y := 3$$

```
> f:='x+y'; g:=x+y;
```

$$f := x + y$$
$$g := 5$$

（注：続けて入力される方はここら辺で restart をかけてください．以下では数値を代入した変数を，free の変数として使っています．そのまま入力を続けると叱られます）

1.2.2 関数

よく使う関数はそのままの形で使えます．三角関数 (trigonometric functions) はラジアンで入れてください．log （もちろん ln も）は自然対数です．底をあらわにするときは

```
> log[2](5);
```

$$\frac{\ln(5)}{\ln(2)}$$

としてください．数値として取り出したいときには

```
> evalf(%);(evaluating float の略)
```

2.321928094

Maple が提供する膨大な数の関数から，目的とするものを探しだすには help を使って下さい．以下に関数等の index を表示する keywords をまとめておきます．

? inifcns - 起動時から認識されている関数

? index[package] - 関連する関数を集めたパッケージです．微分方程式 (DETools), 線形代数 (linalg), プロット関係の関数 (plots) 等があります．

? index[function] - Maple の標準関数．

これらの関数は起動時から読み込まれている関数と，ユーザーが呼び出さなければならない関数とがあります．呼び出しが必要な時には

```
> with( package);
```

でおこなってください．

単純なユーザー関数の定義は次の 2 種類を使います．

i) 矢印による定義.

ii) unapply による定義.

矢印による定義は普通の入力のようにして,

```
> restart:
eq1:=x->exp(-x)*cos(10*x);
```

$$eq1 := x \to e^{(-x)} \cos(10\,x)$$

unapply は一度求めた式を関数として定義するときに使います. たとえば以下では eq1 という式に入っている定義にしたがって, x を変数とする f1 という関数と定義しています.

```
> restart:
eq1:=exp(-x)*cos(10*x);
f1:=unapply(eq1,x);
```

$$eq1 := e^{(-x)} \cos(10\,x)$$

$$f1 := x \to e^{(-x)} \cos(10\,x)$$

まちがって数式に対して矢印での定義をすると

```
> f1:=x->eq1;
```

$$f1 := x \to eq1$$

```
> f1(1);
```

$$e^{(-x)} \cos(10\,x)$$

となり変数を変数とみなしてくれません.

ややこしい操作が必要な関数を定義するには第 1.5節で解説する proc を使います.

1.2.3 プロット

複雑な関数も Maple だとすぐに視覚化してくれます. 先程のユーザー定義関数を使って, 関数が実際にどのような形をしているか plot させてみましょう.

```
> plot(eq1(x),x=0..10);
```



となります．パラメーターによるプロットも可能です．例えば

```
> plot([sin(t),cos(t),t=-Pi..Pi]);
```



によって円が描けます．真円にしたいときには出力図形をマウスで選んだ後，メニューバー下段の (1:1) ボタンを押してください．2 変数の場合には plot3d を使います．変数名が明らかな時には省略可能です．

```
> f1:=(x,y)->sin(x)*cos(y);
> plot3d(f1,-Pi..Pi,-Pi..Pi);
```

$$f1 := (x,\, y) \to \sin(x)\cos(y)$$



オプションや特殊な plotting 法があります．plot に対する一部の簡単な操作（視点の変更，表面の加工，軸の挿入) はメニューバーからできます．さらに

```
> with(plots):
```

で呼び出される plots package には多くの便利な表示関数が用意されています．その一部の機能については後ほど実例をお見せします．詳しい内容はそれぞれの help を参照ください．

1.2.4 方程式の解

solve で方程式の解が求まります．

```
> eqset:={x+y=1,y=1+x^2};
```

$$eqset := \{x + y = 1,\ y = 1 + x^2\}$$

```
> solve(eqset,{x,y});
```

$$\{x = 0,\ y = 1\},\ \{x = -1,\ y = 2\}$$

これだけでは

```
> x;y;
```

$$x$$

$$y$$

のように変数 x,y へ代入されていません．これを代入するには

```
> solset:=solve(eqset,{x,y});
```

$$solset := \{x = 0,\ y = 1\},\ \{x = -1,\ y = 2\}$$

```
> solset[1];
```

$$\{x = 0, y = 1\}$$

```
> assign(solset[1]);
> x;y;
```

$$0$$

$$1$$

のように assign を使います．解を整数の範囲で求める isolve というのもあります（実践演習 [2.1] 参照）．代数的に解けない方程式に対しても数値的に解く関数 (fsolve) がありますが，これについては実践演習 [2.7] で扱っています．

1.2.5 式の変形

式の展開や簡単化のために使われるコマンドを以下にまとめて記します．詳しい意味や使用例はヘルプを参照してください．具体的な使用法は実践演習 [2.2] に示しました．”MapleV ラーニングガイド ”の 2.6 数式の変形 に分かりやすい実例が載っています．

- expand - Expand an Expression （展開）
- simplify - Apply Simplification Rules to an Expression （簡単化，これですべてうまく行くといいのですが...）
- collect - Collect Coefficients of Like Powers（項の次数で式をまとめる）
- combine - combine terms into a single term （規則にしたがって式をまとめる）
- coeff - extract a coefficient of a polynomial（多項式の係数の取り出し）
- sort - sort a list of values or a polynomial （式や値のソート）
- factor - Factor a Multivariate Polynomial （一つ以上の変数を持つ多項式の因数分解）
- normal - normalize a rational expression （約分，通分）
- convert - convert an expression to a different form （関数を違う関数へ変換）
- gcd - greatest common divisor of polynomials （多項式の最大公約数）
- lcm - least common multiple of polynomials （多項式の最小公倍数）

- numer - numerator of an expression （分母）
- denom - denominator of an expression （分子）
- radsimp - simplification of an expression containing radicals （分母の有理化)
- assume - The Assume Facility （変数の範囲や関係を規定）

1.3 微積分とその応用

前章の簡単な数式処理の続きで微分・積分とその周辺の題材です．級数は物理や化学の論文でよく使われています．基本的にはうまく解析的に解が出てこないところで，近似として使われます．したがって，どの程度の近似かを常に意識する必要があります．

1.3.1 総和の計算

総和は

```
> sum(i,i=1..10);
```

$$55$$

のようにして求まります．以下は高校でお目にかかった公式です．

```
> sum(i^2,i=1..n);
```

$$\frac{1}{3}(n+1)^3-\frac{1}{2}(n+1)^2+\frac{1}{6}n+\frac{1}{6}$$

```
> assume(abs(a)<1);
> sum(a^i,i=1..infinity);
> a:='a':
```

$$-\frac{\tilde{a}}{\tilde{a}-1}$$

a は収束するように仮定 (assume) して求めています．あとに残らないよう直後に a を default に戻しています．version 6 ではユーザーが当然と考える assume を自動的に施すようになっていますので，このような操作は不要です．

1.3.2 極限

極限の値は limit によってもとまります．

```
> restart;
> limit(sin(x)/x,x=0);
```

$$1$$

```
> limit(1/x,x=0,complex);
```

$$\infty-\infty I$$

高校で習った知識がそのまま使えます．

```
> sum(1^i/i!,i=0..infinity);
```

$$e$$

例えば，以下のような式に単純な代入を行うと

```
> r:=(x^2-1)/((x+1)*(x^2+2));
```

$$r:=\frac{x^2-1}{(x+1)(x^2+2)}$$

```
> subs(x=-1,r);
Error, division by zero
```

とエラーを返してきます．これの極限をとると

```
> limit(r,x=-1);
```

$$\frac{-2}{3}$$

つぎに右側極限と左側極限とが違う値に収束している場合は

```
> limit(tan(x),x=Pi/2);
```

$$undefined$$

です．しかし，右と左を指定すると

```
> limit(tan(x),x=Pi/2,right);
> limit(tan(x),x=Pi/2,left);
```

$$-\infty$$

$$\infty$$

とちゃんと求めてくれます．

1.3.3 微分

微分は diff によって行います．

```
> diff(x^2,x);
```

$$2\,x$$

```
> diff(y^2*x^2,x,x);
```

$$2\,y^2$$

関数の一般形のままでも形式的な微分を表示してくれます．

```
> c:=(x,t)->X(x)*T(t);
```

$$c := (x,\, t) \to \mathrm{X}(x)\,\mathrm{T}(t)$$

```
> diff(c(x,t),t);
> diff(c(x,t),x,x);
```

$$\mathrm{X}(x)\,(\tfrac{\partial}{\partial t}\,\mathrm{T}(t))$$

$$(\tfrac{\partial^2}{\partial x^2}\,\mathrm{X}(x))\,\mathrm{T}(t)$$

1.3.4 微分方程式

微分方程式は dsolve によって解きます．例えば，

```
> eq:=diff(y(x),x)+y(x)=0;
```

$$eq := (\tfrac{\partial}{\partial x}\,\mathrm{y}(x)) + \mathrm{y}(x) = 0$$

```
> dsolve(eq,y(x));
```

$$\mathrm{y}(x) = \_C1\; e^{(-x)}$$

と求められます．さらに初期条件を付け加えるときには

```
> dsolve({eq,y(0)=a},y(x));
```

$$y(x) = a\,e^{(-x)}$$

として代入します．この他に連立微分方程式，べき級数解 (option=series)，数値解 (option=numeric) 等も求めることができます．連立微分方程式については実践演習 [2.4] に示しました．

1.3.5 積分

積分は，不定積分，定積分はそれぞれ

```
> int(ln(x),x);
```

$$x\ln(x) - x$$

```
> int(sin(x),x=-Pi..0);
```

$$-2$$

などで求めます．int を integrate としても同じ結果を得ます．積分公式がないと求められないような関数も

```
> eq:=x^2/sqrt(1-x^2);
> int(eq,x);
```

$$eq := \frac{x^2}{\sqrt{1-x^2}}$$

$$-\frac{1}{2}\,x\sqrt{1-x^2} + \frac{1}{2}\arcsin(x)$$

```
> eq2:=exp(-x^2);
> int(eq2,x=-z..z);
```

$$eq2 := e^{(-x^2)}$$

$$\sqrt{\pi}\,\mathrm{erf}(z)$$

という具合です．積分を実行するのではなく，積分記号だけを表示させたいときには Int とします．

1.3.6 級数

Maple で式の Tayler 級数展開を見てみましょう．位数は最後につけます．例えばある関数を原点の周りで４次まで展開してみましょう．

```
> series(f(x),x=0,4);
```

$$\mathrm{f}(0) + \mathrm{D}(f)(0)\,x + \frac{1}{2}\,(\mathrm{D}^{(2)})(f)(0)\,x^2 + \frac{1}{6}\,(\mathrm{D}^{(3)})(f)(0)\,x^3 + \mathrm{O}(x^4)$$

この関数を取り出すためには convert を使います．

```
> convert(%,polynom);
```

$$\mathrm{f}(0) + \mathrm{D}(f)(0)\,x + \frac{1}{2}\,(\mathrm{D}^{(2)})(f)(0)\,x^2 + \frac{1}{6}\,(\mathrm{D}^{(3)})(f)(0)\,x^3$$

これと unapply を組み合わせれば，多様な関数の定義ができます．これらの使い方は実践演習 [2.3] に簡単な例を示しました．Maple はさらに漸近級数を求めるために asympt という関数も用意しています．

1.3.7 複素関数

多くの関数は複素数を引数としてとることができます．

```
> series(exp(x),x=0.5*I,3);
```

$$.8775825619 + .4794255386\,I + (.8775825619 + .4794255386\,I)\,(x - .5\,I) + (.4387912810 + .2397127693\,I)\,(x - .5\,I)^2 + \mathrm{O}((x - .5\,I)^3)$$

```
> int(exp(x/2),x=1+1*I..0+1*I);
> evalc(%);
```

$$-2\,e^{(1/2+1/2\,I)} + 2\,e^{(1/2\,I)}$$

$$-2\,e^{(1/2)}\cos(\frac{1}{2}) + 2\cos(\frac{1}{2}) + I\,(-2\,e^{(1/2)}\sin(\frac{1}{2}) + 2\sin(\frac{1}{2}))$$

evalc は複素数の実数部と虚数部をあらわな形に分けて表示してくれます．実数部だけを取りだすときには Re()，虚数部だけは Im()，複素共役を求めるのには conjugate() です．さらに複素平面での関数のプロットもお任せでやってくれます．

```
> with(plots):
> complexplot3d( sec(z) , z = -2 - 2*I ..  2 + 2*I );
```

```
Warning, the name changecoords has been redefined
```



```
> complexplot(sin(x+I),x=-Pi..Pi);
```


1
0.5
–1.5
–1
–0.5
0.5
1
1.5
–0.5
–1

1.4 データ構造と線形代数

Maple は種々のデータ構造を持っています．これらデータ構造の中で混乱しがちな集合，リスト，表，配列についてまとめて説明します．さらに工学で頻繁に使う行列とベクトルを扱う線形代数のパッケージについて述べます．

1.4.1 集合，リスト，表，配列

集合 set {a,b,c}

集合は普通に使われる集合と同じ意味を持ちます．集合に対する組み込みのオペレーターには以下のものがあります．

- FUNCTION:union - set union operator （和）
- FUNCTION:intersect - set intersection operator （積）
- FUNCTION:minus - set difference operator （差）

```
> set1:={1,2,3};
set2:={3,4};
```

$$set1 := \{1, 2, 3\}$$

$$set2 := \{3, 4\}$$

```
> set3:=set1 union set2;
```

$$set3 := \{1, 2, 3, 4\}$$

```
> set4:=set1 intersect set2;
set5:=set1 minus set2;
```

$$set4 := \{3\}$$

$$set5 := \{1, 2\}$$

集合では値が重複する時には一つだけを残して他は削除されます．値には基本的に順番がありません．

リスト list [a,b,c]

リストには順番があります．また，重複する場合にもそのまま残されます．ここではリストの操作について例を挙げます．

```
> list1:=[1,2,3];
> list2:=[3,4];
```

$$list1 := [1, 2, 3]$$

$$list2 := [3, 4]$$

この結合をつくるときには op(list) を使います．この関数は list の要素からなる部分列を生成します．これを使ってリストの内容を取りだし，さらに結合を作ることができます．

```
> list3:=[op(list1),op(list2)];
```

$$list3 := [1, 2, 3, 3, 4]$$

集合と違って，要素は全て保存されます．リストに要素を追加する (append, prepend) ときにも同様の手を使います．リストに含まれている要素の個数を知りたいときには nops を使います．

```
> n:=nops(list3);
```

$$n := 5$$

要素の内容は list[index] で取り出せます．

```
> list3[5];
```

$$4$$

index は 1 から始まります．全ての要素を表示したいときには print(list) を使います．また seq を使っても可能です．これを利用して，逆順に並び替えてみましょう．

```
> list4:=[seq(list3[n-i+1],i=1..n)];
```

$$list4 := [4, 3, 3, 2, 1]$$

並び替えの結果を直接 list3 にするとデータは破壊されます．従って，元の名前を使いたいときにはその後で代入をします．

```
> list3:=list4;
```

$$list3 := [4, 3, 3, 2, 1]$$

データの入れ換えは単純に，

```
> list3[2]:=x;
```

$$list3_2 := x$$

です．全部を表示してみると

```
> list3;
```

$$[4, x, 3, 2, 1]$$

表 table

Maple では表形式のデータ構造を取ることができます．その場合 index(subscript) はアルファベットの名前なども使えます．表を作るときには table 関数を用いて以下のようにします．

```
> atomicweight:=table([Fe=56,C=12]);
```

$$atomicweight := \mathrm{table}([C = 12, Fe = 56])$$

```
> atomicweight[Fe];
```

$$56$$

表の変更は

```
> atomicweight[Fe]:=55.847;
```

$$atomicweight_{Fe} := 55.847$$

で行います．また追加は

```
> atomicweight[Al]:=27;
```

$$atomicweight_{Al} := 27$$

です．内容の表示は print で行います．

```
> print(atomicweight);
```

$$\text{table}([Al = 27,\ C = 12,\ Fe = 55.847])$$

削除は

```
> atomicweight[C]:='atomicweight[C]';
```

$$atomicweight_C := atomicweight_C$$

```
> print(atomicweight);
```

$$\text{table}([Al = 27,\ Fe = 55.847])$$

です．(”ちょん ”はここでは普通のシングルクォート”’”を使います．バックォート”‘”ではありません）

配列 array

配列は表の subscript （添字) を整数に限定したものです．これは行列やベクトルとして使えます．表の場合の table と同じように配列では array を用います．ただし，subscript の範囲を明示する必要があります．ベクトルは array(1..n) で配列は array(1..m,1..n) として指定します．特別の関係を持った行列は以下のようにして定義することができます．

```
> array(identity,1..3,1..3);
> array(sparse,1..3,1..3);
```

$$\begin{bmatrix} 1 & 0 & 0 \\ 0 & 1 & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

$$\begin{bmatrix} 0 & 0 & 0 \\ 0 & 0 & 0 \\ 0 & 0 & 0 \end{bmatrix}$$

```
> array(1..2,1..2,[[1,2],[3,4]]);
```

$$\begin{bmatrix} 1 & 2 \\ 3 & 4 \end{bmatrix}$$

```
> A:=array(antisymmetric,1..3,1..3);
> A[1,2]:=1;A[1,3]:=2;A[2,3]:=3;
> print(A);
```

$$A := \text{array}(antisymmetric,\ 1..3,\ 1..3,\ [])$$

$$A_{1,2} := 1$$
$$A_{1,3} := 2$$
$$A_{2,3} := 3$$
$$\begin{bmatrix} 0 & 1 & 2 \\ -1 & 0 & 3 \\ -2 & -3 & 0 \end{bmatrix}$$

構造の確認，構造変換と関数の適用

type 前に作ったデータがどういう構造を持っているのか分からなくなってしまうときがあります．例えば，リストなのか，配列なのかという場合です．このようなときには

```
> type(A,array);
```

$$true$$

などで確認することができます．返答は true か false です．

convert 集合，リスト，表，配列の相互のデータ構造間の構造変換も行ってくれます．以下では配列をリストのリスト (listlist) に変換しています．

```
> convert(A,listlist);
```

$$[[0,\ 1,\ 2],\ [-1,\ 0,\ 3],\ [-2,\ -3,\ 0]]$$

map 全ての要素に関数を適用したい場合には

```
> map(sin,A);
```

$$\begin{bmatrix} 0 & \sin(1) & \sin(2) \\ -\sin(1) & 0 & \sin(3) \\ -\sin(2) & -\sin(3) & 0 \end{bmatrix}$$

のように map によっておこないます．さらに map によってデータの任意の列から成るあらたな listlist をつくる事も可能です．

```
> map(u->[u[1],u[3]],convert(A,listlist));
```

$$[[0,\ 2],\ [-1,\ 3],\ [-2,\ 0]]$$

1.4.2 線形代数

Maple の標準関数を使って線形代数 (linear algebra) を適用することは可能ですが面倒です．従来は Maple ライブラリーに用意されている linalg package を使っていました．バージョン 6 で新たに LinearAlgebra(LA) が追加され，その速度と操作性の良さから，今後これが標準となるようです．抽象的あるいは厳密な線形代数の計算が必要な時には linalg をお使いください．本書で

は LinearAlgebra にしたがって説明します．linalg の対応する記述は付録 A.4 にまとめています．

Matrix や Vector の生成には以下のようにたくさんの方法があります．

```
> with(LinearAlgebra):
> ma0:=Matrix(A);
> ma1:=<<1,2,3>|<4,5,6>|<7,8,9>>;
```

$$ma0 := \begin{bmatrix} 0 & 1 & 2 \\ -1 & 0 & 3 \\ -2 & -3 & 0 \end{bmatrix}$$

$$ma1 := \begin{bmatrix} 1 & 4 & 7 \\ 2 & 5 & 8 \\ 3 & 6 & 9 \end{bmatrix}$$

```
> vec1:=Vector([x,y,z]);
> vec2:=<1,2,3>;
> vec3:=<1|2|3>;
```

$$vec1 := \begin{bmatrix} x \\ y \\ z \end{bmatrix}$$

$$vec2 := \begin{bmatrix} 1 \\ 2 \\ 3 \end{bmatrix}$$

$$vec3 := [1, 2, 3]$$

等です．ここで Vector は縦（列）ベクトル (column vector) を生成することに注意ください．行列の転置を行う演算 Transpose(vec) を行うと横（行）ベクトル (row vector) となります．(英語で座席は row で，新聞の囲み記事は column です)．

多くの高機能な演算が LinearAlgebra package に用意されています．使い方は以下の通りです．matrix 計算の時には次元が整合していなければなりません．

```
> ma0.ma0;
> MatrixAdd(ma0,ma1,X,Y);
```

$$\begin{bmatrix} -5 & -6 & 3 \\ -6 & -10 & -2 \\ 3 & -2 & -13 \end{bmatrix}$$

$$\begin{bmatrix} Y & X+4\,Y & 2\,X+7\,Y \\ -X+2\,Y & 5\,Y & 3\,X+8\,Y \\ -2\,X+3\,Y & -3\,X+6\,Y & 9\,Y \end{bmatrix}$$

```
> ma0.vec1;
```

$$\begin{bmatrix} y+2\,z \\ -x+3\,z \\ -2\,x-3\,y \end{bmatrix}$$

```
> Transpose(vec1).vec1;
> Multiply(vec1,Transpose(vec1));
```

$$x^2+y^2+z^2$$

$$\begin{bmatrix} x^2 & x\,y & x\,z \\ x\,y & y^2 & y\,z \\ x\,z & y\,z & z^2 \end{bmatrix}$$

全ての演算が with(LinearAlgebra) を実行したときに表示されますので，関連している項目のヘルプを参照してください．以下に主に使う演算を示します．

- MatrixAdd - matrix or vector addition（和）(MatrixAdd(A,B,c1,c2):c1*A+c2*B)
- DotProduct - vector dot (scalar)product （内積）
- CrossProduct - compute the cross product of two Vectors （外積）
- MatrixInverse - compute the inverse of a matrix （逆行列）
- Determinant - determinant of a matrix （行列式）
- Trace - the trace of a matrix （対角和）
- Adjoint - compute the adjoint of a matrix （随伴）
- Transpose - compute the transpose of a matrix （転置行列）
- Eigenvalues - compute the eigenvalues of a matrix （固有値）
- Eigenvectors - find the eigenvectors of a matrix （固有ベクトル）
- VectorAngle - compute the angle between vectors （角度）
- Dimension - determine the dimension of a Matrix or a Vector （要素数）

linalg にもこれらに対応する関数が用意されています．関数名は微妙に違います．linalg の対応する記述は付録 A.4 にまとめています．さらに，linalg にはこの他にも curl,diverge,grad,jacobian などの微分演算子や，行列やベクトルの行や列を操作するオペレーションが用意されています．LinearAlgebra の使用法は第 1.6.4節）と実践演習 [2.5][2.6] に示しました．

1.5 MapleV でのプログラミング

1.5.1 プログラミングの流れ

ある決まった操作をするときや関数が複雑になってきた場合にプログラミングが必要になってきます．Maple では BASIC や C などの一般的なプログラム言語より圧倒的に楽にプログラミングができます．このプログラミングの便利さは，ちょっとしたルーチン計算のために使えるだけでなく，本格的なプログラムを書く前にそのひな型を検討する際にも多いに役立ちます．

では，Maple ではプログラムはどのようにして書くのでしょうか？その前にプログラミングの一般的な注意を述べておきます．

1. 本当にプログラミングが必要か？
   いくらプログラミングが楽だといってもやはりプログラミングには時間がかかります．従って，先ず本当にその処理や関数をわざわざ書く必要があるのかを考えてください.『あるものは使え！』が原則です．Maple で提供されている多くの関数の中に使える関数があるかを先ず探してみてください．”ライトついてますかー問題発見の人間学 ”ドナルド・Ｃ ・ゴース，ジェラルド・Ｍ ・ワインバーグ著（共立出版１９８７ ）.

2. プログラムの作成
   プログラムが必要であるという結論に達したときにはそれを実際に作っていくしかありません．この作業の手順は普通のプログラムと全く同じです．

   - 入力と出力の決定
   - プログラミング（本当の）
   - デバッグ

   となるでしょう．本格的に問題を解くときにはアルゴリズムやデータ構造を考える必要がありますが，我々が解く必要がある問題はほとんどの場合単純です．従って，素直に手でやる計算を自動化するだけで出来上がります．下手にテクニックを考えるよりも何をしているのかが見るだけで理解できるプログラムを書くように心掛けてください．

3. 汎用性？
   さらに汎用性や，多くの人があるいは何度も使う可能性がある場合には ERROR,type check, ヘルプファイル，ユーザライブラリの作成のような作業を行う必要があります．

1.5.2 Maple の制御構造

プログラムでの構造は以下の３種類だけで，あとはこれらの組み合わせです．

if 場合分け

for 決まった分だけぐるぐる

while あるところまでぐるぐる

この構文を以下に示します.

if *condition1* **then**
statement sequence 1
elif *condition2* **then**
statement sequence 2
else
default statement sequence
fi;

for *i* **from** *start* **by** *change* **to** *finish*
do
statement sequence
od;

while *condition*
do
statement sequence
od;

next,break

for や while- loop の途中で，処理を省きたいときや loop から脱出したいときには next,break をそれぞれ使います．構文は

- **if** *condition* **next**; od まで跳んで次の loop へ

```
> for i from 1 to 10 do
> if (modp(i,2)=0) then next fi;
> printf("%d, ",i);
> od;
```

```
1, 3, 5, 7, 9,
```

- **if** *condition* **break**; 一番近い od の外へ出て loop を脱出

```
> for i from 1 to 10 do
> if (i>5) then break fi;
> printf("%d, ",i);
> od;
```

```
1, 2, 3, 4, 5,
```

ひな型

procedure （手続関数）のひな型です.

name := **proc**(*input sequences*)
local *variable sequence;*
global *variable sequence;*
options *option sequence;*
description *descripiton sequence;*
statement 1;
⋮
statement n;
end;

procedure からの戻り値は最後の statement からの出力です.

1.5.3 プログラミングの実践

それでは学生の成績データの平均と分散を求める procedure を題材に実際のプログラミングの流れを見てみましょう．先ずデータを用意します.

```
> list1:=[48,56,68,72];
```

$$list1 := [48, 56, 68, 72]$$

次にひな型の作成とデータがうまく読み込まれているかの確認をしておきます．改行は shift+enter でおこないます.

```
> stat:=proc(data1)
> print(data1);
> end;
```

$$stat := \mathbf{proc}(data1)\,\mathrm{print}(data1)\,\mathbf{end\ proc}$$

```
> stat(list1);
```

$$[48, 56, 68, 72]$$

次に実際の処理がうまく動くか確かめておきます. [1]

```
> n:=nops(list1);
> ave:=add(list1[i],i=1..n)/n;
> disp:=0;
> for i from 1 to n do
> disp:=disp+(list1[i]-ave)^2;
> od:
> disp:=disp/n;
```

$$n := 4$$

$$ave := 61$$

$$disp := 0$$

[1] for を使った総和は教育的なものです．実際の数値を足しあわせるときには例で示してあるような，add 関数を使ったほうが，はるかに計算速度は速いです.

$$disp := 91$$

出力が多すぎる時は od; を od:とセミコロンをコロンに変えることで抑制することができます．これらを先ほど作ったひな型に加え local 変数を宣言しておきます．local variables は proc 内でのみ使われる変数です．proc 内で外部の変数を参照することも可能です．同じ名前が使われているときには内部変数が優先されます．

```
> stat:=proc(data1::list)
> local n,ave,disp,i;
> if nargs=0 then ERROR("no argument") fi;
> n:=nops(list1);
> ave:=add(list1[i],i=1..n)/n;
> disp:=0;
> for i from 1 to n do
> disp:=disp+(list1[i]-ave)^2;
> od:
> disp:=disp/n;
> printf("Average       =%10.5f\n",ave);
> printf("Variance      =%10.5f\n",disp);
> printf("Standard disp.=%10.5f\n",evalf(sqrt(disp)));
> end;
> stat(list1);
```

$stat := \mathbf{proc}(data1::list)$
$\mathbf{local}\, n,\ ave,\ disp,\ i;$
$\quad \mathbf{if}\, \mathrm{nargs} = 0\, \mathbf{then}\, \mathrm{ERROR}(\text{“no argument”})\, \mathbf{end\ if};$
$\quad n := \mathrm{nops}(list1)\,;$
$\quad ave := \mathrm{add}(list1_i,\ i = 1..n)/n\,;$
$\quad disp := 0\,;$
$\quad \mathbf{for}\, i\, \mathbf{to}\, n\, \mathbf{do}\, disp := disp + (list1_i - ave)^2\, \mathbf{end\ do}\,;$
$\quad disp := disp/n\,;$
$\quad \mathrm{printf}(\text{“Average \quad =\%10.5f \textbackslash n”},\ ave)\,;$
$\quad \mathrm{printf}(\text{“Variance \quad =\%10.5f \textbackslash n”},\ disp)\,;$
$\quad \mathrm{printf}(\text{“Standard disp.=\%10.5f \textbackslash n”},\ \mathrm{evalf}(\mathrm{sqrt}(disp)))$
$\mathbf{end\ proc}$

```
Average       =  61.00000

Variance      =  91.00000

Standard disp.=   9.53939
```

ここで，出力の見栄えをよくするために printf 文を使っています．書式は C 言語でおなじみのものですが，慣れない人は help を参照してください．文字列はダブルクォートで囲まれています．またエラー処理を追加しています．さらに最初の引数を data1::list とすることによって，data1 の type が list であるかどうかを Maple が判断してくれます．

上のプログラム中に nargs という変数が出てきています．proc 内には特別

の意味を持った nargs と args というのがあります．これまた C 言語ではおなじみです．nargs は proc がいくつの input を受け取ったかを表しています．それぞれの input は args[i] によって取りだすことが可能です．もうひとつ proc 内で重要な特殊名 procname があり，これはプロシージャの名前が割り当てられています．

もうひとつの重要なコマンドは RETURN で，それ以降の計算が必要ないときに戻り値を持って procedure を抜ける関数です．これを使った再帰的な関数を定義してみましょう．近似で有用な Chebyshev 多項式 です．これを漸化式で表すと

$$T_n(x) = 2xT_{n-1}(x) - T_{n-2}(x) \quad \text{for n >= 2} \tag{1.1}$$

で，$T_0(x) = 1$ と $T_1(x) = x$ です．これを Maple で表すと，

```
> T:=proc(n::nonnegint,x::name)
> #Chebyshev polynomials
> option remember;
> if n=0 then RETURN(1);
> elif n=1 then RETURN(x);
> fi;
> 2*x*T(n-1,x)-T(n-2,x);
> end;
```

$T := \mathbf{proc}(n::nonnegint,\ x::name)$
$\mathbf{option}\ remember;$
$\quad \mathbf{if}\ n = 0\ \mathbf{then}\ \mathrm{RETURN}(1)\ \mathbf{elif}\ n = 1\ \mathbf{then}\ \mathrm{RETURN}(x)\ \mathbf{end\ if};$
$\quad 2 * x * \mathrm{T}(n-1,\ x) - \mathrm{T}(n-2,\ x)$
$\mathbf{end\ proc}$

となります．option remember は remember table に結果を保存させて計算速度を上げるためにつけてあります．# はコメントに使っています．この結果を plot させてみます．まず，5 次の Chebyshev 多項式までを関数として定義します．“||” は連結作用素で二つの要素を連結してくれます．

```
> for i from 0 to 5 do
> f||i:=unapply(expand(T(i,x)),x);
> od;
```

$$f0 := 1$$
$$f1 := x \to x$$
$$f2 := x \to 2\,x^2 - 1$$
$$f3 := x \to 4\,x^3 - 3\,x$$
$$f4 := x \to 8\,x^4 - 8\,x^2 + 1$$
$$f5 := x \to 16\,x^5 - 20\,x^3 + 5\,x$$

```
> plot({f0,f1,f2,f3,f4,f5},-1..1);
```



“Maple V プログラミングガイド” には proc などの厳密な規則と有益な実例がのっています．ぜひ，一読されることをお奨めします．

1.5.4 その他のテクニック

debug に関する注意

プログラムにはバグが付き物です．バグ取りには printlevel,trace, mint,profile 等を使います．この中で一番簡単なのは計算途中の出力の量を変えてくれる，printlevel の設定です．これは printlevel :=11 などとして，設定します．元に戻すには printlevel:=1 としてください．trace は特定のプロシージャの結果だけを調べます．mint は maple とは別のプログラムとして用意され syntax のチェックなどをしてくれ，本格的なプログラムを作るときなどに便利です．（付録 B.1で使用例を紹介しています）．さらに実行速度が問題になるときには profile を使います．その他，help file,user library の作り方も含めて help あるいは “Maple V プログラミングガイド” を参照してください．

定義関数の plot に関する注意

例えば

$$\text{filter}(x) \quad = \quad \begin{array}{l} 1 - x/10 \quad \text{for x} < 10 \\ 0 \quad \text{for x} >= 10 \end{array} \tag{1.2}$$

という関数を考えてみましょう．これは

```
> filter:=x-> if x<10 then 1-x/10 else 0 fi;
```

$filter := \mathbf{proc}(x)\,\mathbf{option}\,operator,\ arrow;\,\mathbf{if}\,x < 10\,\mathbf{then}\ \ 1 - 1/10 * x\,\mathbf{else}\,0\,\mathbf{end\ if}\mathbf{end\ proc}$

となります．うまく定義できているか確認しておきましょう．

```
> plot('filter(x)',x=0..20);
```



ここで，自分で定義した関数の plot を行うときにはシングルクォートで囲む必要があることに注意ください．plot では初めに引数がチェックされます．したがって，ユーザー定義関数では引数が名前であるために if 文などでうまく処理されず error が返ってきます．これを回避するにはシングルクォートを使って引数の評価を行わずに plot コマンドへ渡すことで解決できます．

1.6 データ処理

Maple の応用として，有用なデータ処理に関連して，

- データの読み込み,
- フーリエ変換,
- 畳み込み，そして
- 非線形の最小二乗法

について解説し，具体的な操作を見ていきます.

（参考 河合潤：『化学計測学』 合 志陽一編 昭 晃堂 1997 ，より高度な Filtering や Fitting の記述が，W.Gander and U.von Matt,"Smoothing Filters",Solving Problems inScientific Computing Using Maple and MATLAB,Walter Gander and Jiri Hrebicek,Springer 1991,Chapter 9. にあります．ただし MATLAB のスクリプトです)

1.6.1 データの入出力

データは例えば,

```
> f1:=t->subs(a=10,b=40000,c=380,d=128,a+b/(c+(t-d)^2) );
> T:=[seq(f1(i)*(0.6+0.8*evalf(rand()/10^12)),i=1..256)]:
> writedata(`DATA101`,convert(T,list));
```

によって生成，ファイルに保存することができます.

手っ取り早くデータを読み込むには readdata で行います.

```
> restart:
> T:=readdata(`DATA101`,1):
```

コロンで出力を抑制しています．ここで，最初の引数がファイルの名前で，次がデータの列の数です．ファイル名は特殊な記号（例えばスラッシュ／）が入ったときにはバッククォート（ ` ）で囲っておく必要があります．絶対パス名を書かなかった場合は Mac では Maple の入っているフォルダーだけを[2]，Windows では Current directory をさします．unix では Maple を起動したディレクトリーだけを探します.

ちゃんと読めているか見ておきましょう.

```
> T[1];
> n:=nops(T);
```

$$11.70159307$$

$$n := 256$$

となっています．浮動小数点のデータも自動的に読まれています．もっと複雑なデータや，コメントなどが入っているデータは stats [importdata]

[2] Mac での絶対パス名は例えば，`Mac HD:Desktop Folder:tmp1`などです.

で行ってください（詳しくはヘルプを参照）．次にこのデータが正しく読み込まれているかをチェックしましょう．一個一個確かめるのも手ですが全体を表示させてその形で判断してみましょう．最も簡単には

```
> with(plots):
> listplot(T);
```

```
Warning, the name changecoords has been redefined
```



でできます．うまく読めているようです．ただこの関数ではあまり自由に加工できません．以下では点の座標の並び (listlist) を使って出力する方法を示しておきます．

```
> datapoint:=[seq([i,T[i]],i=1..256)]:
> plot(datapoint,style=point);
```



1.6.2 フーリエ変換

フーリエ変換を用いたデータのフィルタリングについての例です．Maple では FFT(高速フーリエ変換）のルーチンが用意されています．詳しくは help を参照下さい．

演習

与えられた 256 個のデータ（ファイル名 DATA101 ）をフーリエ変換し，高周波成分を下図の三角フィルタを用いて取り除き，フーリエ逆変換せよ．上図のノイズ成分はどうなったか．振幅強度 $\sqrt{a_n^2 + b_n^2}$ をプロットせよ（128 チャンネル）．はじめのデータの面積強度とフーリエ変換で求めたデータの面積強度とはどういう関係があるのか．



解説

時系列データ x (t) の解析にはそのフーリエ変換

$$X(f) = \int_{-\infty}^{\infty} x(t) \exp(2\pi i f t) dt \tag{1.3}$$

が最も多用されます．これは，高速フーリエ変換アルゴリズムの発明によって計算機での処理が容易となったことが大きく効いています．時間経過と共に変化する時系列測定データ $x(t)$ に含まれるさまざまな周波数成分のすべてについて，何 Hz の周波数成分がどの程度の割合でその時系列データの中に含まれているのかを明らかにできるからです．時系列データの中のノイズは δ 関数に似ており，そのフーリエ変換は全周波数成分を一様に含んでいます．

一方，測定したいデータは一般に，時間に対してゆっくり変化する量です．測定によって得られたデータの時間軸は連続ではなくとびとびの値をとるので，離散フーリエ変換を取ることになります．この例では時間データは 256 チャンネルのデータです．高周波成分（=ノイズ成分）を取り除くために，データに対して窓関数（低域通過フィルタ）をかけてフーリエ逆変換

$$x(t) = \int_{-\infty}^{\infty} X(f) \exp(2\pi i f t) dt \tag{1.4}$$

することによってフィルターを掛けた後のスムーズなカーブが得られます．低域通過フィルタには，方形窓，修正方形窓，フォンハン窓（2 乗余弦窓），ハミング窓（平たん部のある 2 乗余弦窓）などさまざまなものがあります．

実践

具体的に計算を見ていきましょう．

```
> restart:
> T:=readdata(‘DATA101‘,1):
> Idata:=array([seq(0,i=1..256)]):
> Rdata:=convert(T,array):
```

全パワーを求めておきましょう．

```
> temp:=add(i^2,i=T);
```

$$temp := 534310.3363$$

実際に FFT を実行します．

```
> FFT(8,Rdata,Idata);
```

$$256$$

これだけで終わりです．ここで FFT の第一引数の 8 はデータの個数が 2^8 であることを示しています．フーリエ変換された後のデータ構造を見ておきましょう．

```
> printf("%3d %15.5f %15.5f\n",1,Rdata[1],Idata[1]);
> for i from 2 to 128 do
> printf("%3d %15.5f %15.5f %15.5f %15.5f\n",
> i,Rdata[i],Idata[i],Rdata[258-i],Idata[258-i]);
> od;
```

```
  1      8499.54360         0.00000
  2     -4162.42568       -81.92030     -4162.42568        81.92031
  3      2602.79575       120.20850      2602.79575      -120.20850
```

⋮

中略

⋮

```
125      -160.89158      -128.35496      -160.89158       128.35496
126        91.48342        49.80754        91.48342       -49.80754
127        36.31291       -82.64862        36.31291        82.64862
128        53.48855         6.15179        53.48855        -6.15179
```

波数空間での強度を logplot で見てみましょう.

```
> Adata:=[seq([i,sqrt(Idata[i]^2+Rdata[i]^2)],i=1..128)]:
> with(plots):
> logplot(Adata);
```

```
Warning, the name changecoords has been redefined
```



離散的な形の Parseval の定理が成り立っていることも確認できます.

```
> add(Rdata[i]^2+Idata[i]^2,i=1..256)/256;
```

$$534310.3320$$

先程求めた実空間での全パワーの結果と一致しています. 次にフィルター関数を作りましょう.

```
> filter:=x-> piecewise(x>=0 and x<=20,(1-x/20) );
> plot(filter(x),x=0..40,title="FILTER FUNCTION");
```

$$filter := x \rightarrow \mathrm{piecewise}(0 \le x \ \mathbf{and}\ x \le 20,\ 1 - \frac{1}{20}\,x)$$

この結果が本節の初めに示したフィルター関数の図です. フィルターを通したデータを作ります.

```
> FRdata:=array([seq(Rdata[i]*filter(i),i=1..256)]):
> FIdata:=array([seq(Idata[i]*filter(i),i=1..256)]):
```

逆フーリエ変換を実行すると,

```
> iFFT(8,FRdata,FIdata);
> Adata:=[seq([i,FRdata[i]],i=1..256)]:
> plot(Adata);
```



ノイズが取り除かれているのが分かるでしょう.

1.6.3 畳み込み (コンボリューション)

波長 t の真のスペクトルが下図の箱形関数 $x(t)$ のようなかたちをしている. 分光器の装置関数が下図の三角関数 $h(t)$ で表されるとき，観測されるスペクトル

$$f(t) = \int_{-\infty}^{\infty} x(t-\tau)h(\tau)d\tau \tag{1.5}$$

はどのような形状となるか.



解説

時刻 t の観測値 $x(t)$ を観測しようとするとき，その t の瞬間のみではなく，t の過去と未来が重み $h(t)$ で見えることがあります．このとき実測されるスペクトルは $x(t)$ と $h(t)$ の畳み込み (convolution)，

$$f(t) = \int_{-\infty}^{\infty} x(t-\tau)h(\tau)d\tau \tag{1.6}$$

となります．スリットを持つ分光器によって，ある波長 t のスペクトル強度 $x(t)$ を観測するとき，分光器はその前後の波長 $t-\tau$ の光も重み $h(t-\tau)$ で見ていることを意味しています．畳み込みを間単に $x*h$ と表すと，そのフーリエ変換は

$$F(x*h) = F(x)F(h) \tag{1.7}$$

となります．すなわち convolution のフーリエ変換は，フーリエ変換の積になるわけです．目的のスペクトル $x(t)$ を求めるためには，わり算 $F(x*h)/F(h)$ の結果をフーリエ逆変換すればよいことがわかります．convolution の形の積分方程式を解くより，わり算の方が計算上簡単な操作ですむ場合が多く，これを deconvolution といいます．

実践

```
> restart;
> fun:=x->piecewise(x>=0 and x<=1,1);
> h:=x->piecewise(x>=0 and x<=1, 1-x);
> plot({fun,h},-1..2);
```

$$fun := x \to \text{piecewise}(0 \le x \textbf{ and } x \le 1,\ 1)$$

$$h := x \to \text{piecewise}(0 \le x \textbf{ and } x \le 1,\ 1-x)$$

前掲の関数が描かれます．convolution を実際に計算すると

```
> plot(int(fun(x)*h(t-x),x=-1..1),t=-2..4);
```



と求まります.

1.6.4 非線形最小二乗法

非線形の最小二乗法を用いてデータフィットをしてくれる関数は default では Maple には用意されていないようです[3]．そこで，実際に非線形最小二乗法のプログラムを作成し，実際のデータへのフィッティングを試みてみましょう.

問題

与えられたデータ (ファイル名'DATA101'）を,

$$f(x) = a + \frac{b}{c + (x - d)^2} \tag{1.8}$$

なるローレンツ型関数にカーブフィッティングするプログラムを作成し，そのパラメータを決定せよ．ここで a はバックグラウンドの強度，b はローレンツ関数の強度，c は線幅，d はピーク位置を表す.

[3]公式のサポートではありませんが，世界中の科学者，技術者が Maple で開発した library を公開している The Maple Application Center (http://www.mapleapps.com/) の Data Analysis のカテゴリーに，後述する Levenberg-Marguqrdt 法を用いた Data fitting の汎用プログラム Generalized Weighted Non-Linear Regression Using the Levenberg-Marquardt Method by David Holmgren (http://www.mapleapps.com/categories/data_analysis_stats/data/html/genfit_6.html) があります.

解説

(1.8) 式の特徴は，パラメータが線形関係にないということですが，以下では線形近似に基づいた反復解法を用います．非線形最小二乗法の注意事項は補足に記しました．

今，パラメータの初期値を $(a_0+\Delta a_1, b_0+\Delta b_1, c_0+\Delta c_1, d_0+\Delta d_1)$ としましょう．このとき関数 f を真値 (a_0, b_0, c_0, d_0) のまわりでテイラー展開し高次項を無視すると

$$\begin{aligned}\Delta f &= f(a_0+\Delta a_1, b_0+\Delta b_1, c_0+\Delta c_1, d_0+\Delta d_1) \\ &\quad -f(a_0, b_0, c_0, d_0) \\ &= \frac{\partial f(a_0,b_0,c_0,d_0)}{\partial a}\Delta a_1 + \frac{\partial f(a_0,b_0,c_0,d_0)}{\partial b}\Delta b_1 \\ &+ \frac{\partial f(a_0,b_0,c_0,d_0)}{\partial c}\Delta c_1 + \frac{\partial f(a_0,b_0,c_0,d_0)}{\partial d}\Delta d_1\end{aligned} \tag{1.9}$$

となります．(1.8) 式の偏微分は計算可能です．'DATA101' のデータは $t=1$ から 256 までの時刻に対応したデータ点 $f_1 \sim f_{256}$ とします．各測定値とモデル関数から予想される値との差 $\Delta f_1 \sim \Delta f_{256}$ は，

$$\begin{pmatrix} \Delta f_1 \\ \Delta f_2 \\ \vdots \\ \Delta f_{256} \end{pmatrix} = J \begin{pmatrix} \Delta a_1 \\ \Delta b_1 \\ \Delta c_1 \\ \Delta d_1 \end{pmatrix} \tag{1.10}$$

となります．ここで 4 行 256 列の行列

$$J = \begin{pmatrix} \left(\frac{\partial f}{\partial a}\right)_1 & \left(\frac{\partial f}{\partial b}\right)_1 & \left(\frac{\partial f}{\partial c}\right)_1 & \left(\frac{\partial f}{\partial d}\right)_1 \\ \vdots & \vdots & \vdots & \vdots \\ \left(\frac{\partial f}{\partial a}\right)_{256} & \left(\frac{\partial f}{\partial b}\right)_{256} & \left(\frac{\partial f}{\partial c}\right)_{256} & \left(\frac{\partial f}{\partial d}\right)_{256} \end{pmatrix} \tag{1.11}$$

はヤコビ行列と呼ばれる行列です．逆行列 J^{-1} は転置行列 J^t を用いて

$$J^{-1} = (J^t J)^{-1} J^t \tag{1.12}$$

と表されます．従って真値からのずれは

$$\begin{pmatrix} \Delta a_2 \\ \Delta b_2 \\ \Delta c_2 \\ \Delta d_2 \end{pmatrix} = (J^t J)^{-1} J^t \begin{pmatrix} \Delta f_1 \\ \Delta f_2 \\ \vdots \\ \Delta f_{256} \end{pmatrix} \tag{1.13}$$

として計算できます．理想的には $(\Delta a_2, \Delta b_2, \Delta c_2, \Delta d_2)$ は $(\Delta a, \Delta b, \Delta c, \Delta d)$ に一致するはずですが，測定誤差と高次項のために一致しません．初期値に

比べ，より真値に近づくだけです．そこで，新たに得られたパラメータの組を新たな初期値に用いて，より良いパラメータに近付けていくという操作を繰り返します．新たに得られたパラメータと前のパラメータとの差がある誤差以下になったところで計算を打ち切り，フィッティングの終了となります．

実践

実際にパラメータフィットを行っていきましょう．線形代数計算のためにサブパッケージとして LinearAlgebra を呼びだしておきます．

```
> restart;
> with(LinearAlgebra):
```

データを読み込みます．

```
> T:=readdata("DATA101",1):
> datapoint:=[seq([i,T[i]],i=1..256)]:
```

(1.8) 式のローレンツ型の関数を仮定しておきましょう．

```
> f:=a+b/(c+(x-d)^2);
> f1:=unapply(f,x);
```

$$f := a + \frac{b}{c + (x - d)^2}$$

$$f1 := x \to a + \frac{b}{c + (x - d)^2}$$

(1.8) 式の関数の微分を求め，新たな関数として定義します．

```
> dfda:=unapply(diff(f,a),x);
> dfdb:=unapply(diff(f,b),x);
> dfdc:=unapply(diff(f,c),x);
> dfdd:=unapply(diff(f,d),x);
```

$$dfda := 1$$

$$dfdb := x \to \frac{1}{c + (x - d)^2}$$

$$dfdc := x \to -\frac{b}{(c + (x - d)^2)^2}$$

$$dfdd := x \to -\frac{b\,(-2\,x + 2\,d)}{(c + (x - d)^2)^2}$$

初期値を仮定します．

```
> guess1:={a=10,b=1200,c=10,d=125};
> plot([datapoint,subs(guess1,f1(x))],x=1..256);
```

$$guess1 := \{c = 10,\ a = 10,\ d = 125,\ b = 1200\}$$



(1.10) 式の左辺の Δf_i を求めます．データの出力を抑制するために od の後はセミコロンではなくコロンにしてあります．デバッグや慣れないときには出力を多めに，データ数を少なめにして，関数の内側から順番に内容を確認しながら打ち込んで下さい．

```
> imax:=nops(T);
> df:=Vector([seq(evalf(subs(guess1,T[i]-f1(i))),i=1..imax)]);
```

$$imax := 256$$

$df :=$ [256 Element Column Vector Data Type : anything Storage : rectangular Order : Fortran_order]

次に (1.11) 式に従ってヤコビ行列を求めます．

```
> Jac:=Matrix(sparse,1..imax,1..4);
> for i from 1 to imax do
> Jac[i,1]:=evalf(subs(guess1,dfda(i)));
> Jac[i,2]:=evalf(subs(guess1,dfdb(i)));
> Jac[i,3]:=evalf(subs(guess1,dfdc(i)));
> Jac[i,4]:=evalf(subs(guess1,dfdd(i)));
> od:
```

$Jac :=$ [256 x 4 Matrix Data Type : anything Storage : rectangular Order : Fortran_order]

(1.12) 式の公式によるヤコビ行列の逆行列の計算です．

```
> IJac:=MatrixInverse(Multiply(Transpose(Jac),Jac)):
> tJac:=Multiply(IJac,Transpose(Jac)):
```

最後に (1.13) 式により，新たなパラメータの組を求めます．

```
> guess2:=Multiply(tJac,df);
```

$$guess2 := \begin{bmatrix} 12.6937323162751525 \\ 4132.31700686929162 \\ 42.3461242313161322 \\ .616791465529281768 \end{bmatrix}$$

```
> guess3:={a=subs(guess1,a)+guess2[1],
> b=subs(guess1,b)+guess2[2],
> c=subs(guess1,c)+guess2[3],
> d=subs(guess1,d)+guess2[4]};
```

$guess3 := \{a = 22.69373232,\ d = 125.6167915,\ b = 5332.317007,\ c = 52.34612423\}$

求めたパラメータを用いたモデル関数と，データをプロットしてみます.
前回より近づいているのがわかるでしょう.

```
> plot({datapoint,subs(guess3,f1(x))},x=1..256);
```




```
> guess1:=guess3:
```

として新たな初期値とし，Δf_i 以下の操作を数回繰り返し，誤差をあらわす guess2 の値がすべて 10^{-5} 以下になった後の結果です．見やすくするために出力を少し工夫してあります.

```
> print(guess1);
```

$\{c = 353.4305681,\ a = 9.976958126,\ b = 39219.46517,\ d = 128.7272859\}$

```
> with(plots):
> F2:=plot({datapoint},x=1..256,color=red,style=point):
> F1:=plot(subs(guess1,f1(x)),x=1..256,style=line,color=blue,thickness=1
> ):
> display({F1,F2},title="Raw data and fitted curve");
```

```
Warning, the name changecoords has been redefined
```



補足:最小二乗法に関するメモ

前述のフィッティング法の説明では，テイラー展開を用いた説明であり，まるで最小二乗法を用いてないような印象を与えたかもしれません．しかしこれは，最小二乗法の χ^2 関数に Newton-Raphson 法を適用し，二次微分を無視した方法と考えることも可能です．この様子を以下に記しておきます．

当てはめたいモデルは x がデータの横軸，$\mathbf{a}$ がパラメータの組とすると，

$$y = f(x;\mathbf{a}) \tag{1.14}$$

です．最小二乗法の χ^2 関数は

$$\chi^2(\mathbf{a}) = S(\mathbf{a}) = \sum_{i=1}^{N} [y_i - f(x_i;\mathbf{a})]^2 \tag{1.15}$$

です．後の式を単純にするために $y_i - f(x_i;\mathbf{a}) \Rightarrow f(x_i;\mathbf{a})$ と置き換え，あらかじめ実験値を引いておきます．S が $\mathbf{a}$ で極小値を持つ (安定である) ための条件は

$$\frac{\partial S(\mathbf{a})}{\partial a_k} = -2\sum_{i=1}^{N} f(x_i;\mathbf{a})\frac{\partial f(x_i;\mathbf{a})}{\partial a_k} = 0. \tag{1.16}$$

ヤコビ行列を $J_i(\mathbf{a}) = \left[\frac{\partial f_i}{\partial a_j}\right]$ と表すと

$$\nabla S(\mathbf{a}) = -2J^t(\mathbf{a})f(y;\mathbf{a}) = 0 \tag{1.17}$$

と書き換えることができます．これに Newton-Raphson 法を適用すると

$$\nabla^2 S(\mathbf{a})\Delta\mathbf{a}^{(k)} = -\nabla S(\mathbf{a}^{(k)}) \tag{1.18}$$

となります． $S(\mathbf{a})$ のあらわな 2 次微分の表現は

$$\frac{\partial^2 S(\mathbf{a})}{\partial a_k \partial a_{k'}} = -2\left\{\sum_{i=1}^{N} \frac{\partial f(x_i;\mathbf{a})}{\partial a_k}\frac{\partial f(x_i;\mathbf{a})}{\partial a_{k'}} + \sum_{i=1}^{N} \frac{\partial^2 f(x_i;\mathbf{a})}{\partial a_k \partial a_{k'}} f(x_i;\mathbf{a})\right\} \quad (1.19)$$

です．ここで，k 回目の推定パラメータ $\mathbf{a}^{(k)}$ に対する修正値を $\Delta\mathbf{a}^{(k)}$ とすると，

$$\left\{J^t\left(\mathbf{a}^{(k)}\right) J\left(\mathbf{a}^{(k)}\right) + \sum_{i=1}^{N} f\left(x_i;\mathbf{a}^{(k)}\right) \nabla^2 f\left(x_i;\mathbf{a}^{(k)}\right)\right\} \Delta\mathbf{a}^{(k)} = J^t\left(\mathbf{a}^{(k)}\right) f\left(x_i;\mathbf{a}^{(k)}\right) \quad (1.20)$$

が得られます．鉤括弧内の第 2 項を無視すると，先程の（1.10）式の線形的な関係が現れます．この無視する項は，線形の場合には確かに現れないし，1 次の項に比べて小さいと考えられます．さらに $f(x_i;\mathbf{a})$ を掛けて和を取っているため，それらがお互いにキャンセルしてくれる可能性があります．

この Gauss- Newton 法と呼ばれる非線形最小二乗法は線形問題から拡張した方法として論理的に簡明であり，広く使われています．しかし，収束性は高くなく，むしろ発散しやすいので注意が必要です．先程の 2 次の項を無視するのでなく，うまく見積もる方法を用いたのが Levenberg-Marquardt 法です．明快な解説が Numerical Recipes in C(C 言語による数値計算のレシピ) WilliamH.Press 他著，技術評論社 1993 にあります．

第2章　実践演習

うまく結果が出ないときに試してください.
以下の入力を打ち込むときの一般的なコツを最初にまとめておきます.

1. restart をかける：続けて入力すると前の入力が生きています．違う問題へ移るときやもう一度入力をし直すときには restart を入力してください.

2. 内側から順に入力する：長い入力は内側の関数から順に何をしているか確認しながら打ち込んで下さい.

3. 出力してみる：紙面の関係で出力を抑止していますが，できるだけ多く出力するようにしてください．最後の：を；に変えるとか，printf を使ってください.

4. 関数に値を代入してみる：数値が返ってくるべき時に変数があればどこかで入力をミスっています．plot で Plotting error,empty plot が出た場合にチェック下さい.

2.1 化学反応式の係数決定（連立方程式の整数解）

次の反応式

$$a\mathrm{MnO_4^{2-}} + b\mathrm{H^+} \rightarrow c\mathrm{MnO_4^-} + d\mathrm{MnO_2} + e\mathrm{H_2O} \tag{2.1}$$

の係数を元素と電荷の保存から求めよ.（参考『はじめての Maple 』，B.W. チャー他著，サイバネットシステム株訳，シュープリンガー・フェアラーク東京，1993,p.205）

実践

先ず元素，電荷の保存の条件式を代入します.

```
> Mn:=a=c+d;
> OO:=4*a=4*c+2*d+e;
> H:=b=2*e;
> Q:=-2*a+b=-c;
```

$$Mn := a = c + d$$

$$OO := 4\,a = 4\,c + 2\,d + e$$

$$H := b = 2\,e$$

$$Q := -2\,a + b = -c$$

これを整数の範囲で連立方程式を解いてくれる isolve を使って解きます.

```
> ans:=isolve({Mn,OO,H,Q});
```

$$ans := \{a = 3\,\_Z1,\ b = 4\,\_Z1,\ c = 2\,\_Z1,\ d = \_Z1,\ e = 2\,\_Z1\}$$

方程式と未知数の数が一致していません．従って任意定数 $\_Z1$ が出てきます．$\_Z1$ に 1 を代入してそれぞれの未知変数を決めます.

```
> subs(_Z1=1,ans);
```

$$\{a = 3,\ b = 4,\ c = 2,\ d = 1,\ e = 2\}$$

2.2 トンネル効果（式の変形）

（参考：シッフ著 量子力学（井上 健訳），吉岡書店 1970．小出昭一郎著 基礎物理学選書 5A ー量子力学，裳華房 1969．



基礎

量子効果の基礎的な例である 1 次元のトンネル効果についての式を導いてみましょう．上図のようなポテンシャルを考えます．詳しい解説は成書を参考にしていただくとして，以下で扱う数式の簡単な説明だけをしておきます．まず，ポテンシャルエネルギー $V(x) = 0$ の領域での波動方程式は

$$-\frac{\hbar^2}{2m}\frac{d^2\varphi(x)}{dx^2} = \varepsilon\varphi(x) \tag{2.2}$$

ですから，波動関数は

$$\begin{aligned} x \le 0 \quad &\text{では} \quad \varphi(x) = A\exp(ikx) + B\exp(-ikx) \\ x \ge a \quad &\text{では} \quad \varphi(x) = C\exp(ikx). \end{aligned} \tag{2.3}$$

ここで $k = \sqrt{2m\varepsilon/\hbar^2}$ は波数ベクトルの大きさです．

ポテンシャルの壁の内側では $\varepsilon \lesseqgtr V_0$ によって事情が変わってきます．$\varepsilon \ge V_0$ では $\kappa = \sqrt{2m(\varepsilon - V_0)/\hbar^2}$ と定義すると，波動関数は

$$0 \le x \le a \text{ では } \varphi(x) = F\exp(i\kappa x) + G\exp(-i\kappa x) \tag{2.4}$$

となります．波動関数は粒子の座標に関する滑らかな連続関数でなければな

らないという条件を $x=0$ と $x=a$ に適用すると，条件はそれぞれ

$$\begin{aligned}
&x=0 \quad \text{で } \varphi(x) \text{ が連続：} \quad A+B=F+G \\
&x=0 \quad \text{で } \varphi'(x) \text{ が連続：} \quad k(A-B)=\kappa(F-G) \\
&x=a \quad \text{で } \varphi(x) \text{ が連続：} \quad F\exp(i\kappa a)+G\exp(-i\kappa a)=C\exp(ika) \\
&x=a \quad \text{で } \varphi'(x) \text{ が連続：} \quad \kappa F\exp(i\kappa a)-\kappa G\exp(-i\kappa a)=kC\exp(ika)
\end{aligned} \tag{2.5}$$

で与えられます．係数が 5 個で，方程式が 4 個ですから，それぞれの係数の比だけが求まります．これらから F,G を消去して，B/A 入射波と反射波の複素振幅の比，および C/A 入射波と透過波の複素振幅の比が求まります．これらの二乗が反射係数と透過係数にほかなりません．結果は

$$\begin{aligned}
\left|\frac{B}{A}\right|^2 &= \left[1+\frac{4k^2\kappa^2}{(k^2-\kappa^2)^2\sin^2\kappa a}\right]^{-1} = \left[1+\frac{4\varepsilon(\varepsilon-V_0)}{V_0^2\sin^2\kappa a}\right]^{-1} \\
\left|\frac{C}{A}\right|^2 &= \left[1+\frac{(k^2-\kappa^2)^2\sin^2\kappa a}{4k^2\kappa^2}\right]^{-1} = \left[1+\frac{V_0^2\sin^2\kappa a}{4\varepsilon(\varepsilon-V_0)}\right]^{-1}
\end{aligned} \tag{2.6}$$

となります．

$0<\varepsilon<V_0$ では $\alpha=\sqrt{2\varepsilon(V_0-\varepsilon)/\hbar^2}$ として，波動関数は

$$0\le x\le a \text{ では} \varphi(x)=F\exp(\alpha x)+G\exp(-\alpha x) \tag{2.7}$$

です．同様な計算によって

$$\left|\frac{C}{A}\right|^2 = \left[1+\frac{V_0^2\sinh^2\alpha a}{4\varepsilon(\varepsilon-V_0)}\right]^{-1} \tag{2.8}$$

となります．$mV_0a^2/\hbar^2=8$ の場合の透過率を求めると下図のようになります．$|E/V_0|<1$ でも透過波の比強度 $|C/A|^2$ が有限の値を取る現象がトンネル効果です．



実践

境界条件から複素振幅の比の導出を実際に Maple で行ってみましょう.
先ず境界での連続条件を打ち込むと,

```
> restart;
> eq1:=A+B=F+G;
> eq2:=k*(A-B)=kappa*(F-G);
> eq3:=F*exp(I*kappa*a)+G*exp(-I*kappa*a)=C*exp(I*k*a);
> eq4:=kappa*F*exp(I*kappa*a)-kappa*G*exp(-I*kappa*a)=k*C*exp(I*k*a);
```

$$eq1 := A + B = F + G$$

$$eq2 := k\,(A - B) = \kappa\,(F - G)$$

$$eq3 := F\,e^{(I\,\kappa\,a)} + G\,e^{(-I\,\kappa\,a)} = C\,e^{(I\,k\,a)}$$

$$eq4 := \kappa\,F\,e^{(I\,\kappa\,a)} - \kappa\,G\,e^{(-I\,\kappa\,a)} = k\,C\,e^{(I\,k\,a)}$$

次に B,C を求めてみましょう.

```
> solve({eq1,eq2,eq3,eq4},{B/A,C/A});
```

としても何も返ってきません．Maple はこれだけではお手上げな様です.
仕方がないので，順番に解いていきましょう．F,G を消去するという方針
にしたがって,

```
> sol1:=solve({eq3,eq4},{F,G});
```

$$sol1 := \{G = \frac{1}{2}\,\frac{C\,(\kappa - k)\,e^{(I\,\kappa\,a + I\,k\,a)}}{\kappa},\ F = \frac{1}{2}\,\frac{C\,(\kappa + k)\,e^{(I\,k\,a - I\,\kappa\,a)}}{\kappa}\}$$

アサインで F,G へ答えを代入することを忘れずに !!

```
> assign(sol1);
> F;
```

$$\frac{1}{2}\,\frac{C\,(\kappa + k)\,e^{(I\,k\,a - I\,\kappa\,a)}}{\kappa}$$

```
> sol2:=solve({eq1,eq2},{A,B});
```

$$sol2 := \{A = \frac{1}{4}\,\frac{C\,(-\%1\,\kappa^2 + \%2\,\kappa^2 + \%2\,k^2 - \%1\,k^2 + 2\,k\,\%2\,\kappa + 2\,k\,\%1\,\kappa)}{k\,\kappa},$$

$$B = -\frac{1}{4}\,\frac{C\,(\%2\,\kappa^2 - \%2\,k^2 + \%1\,k^2 - \%1\,\kappa^2)}{k\,\kappa}\}$$

$$\%1 := e^{(I\,\kappa\,a + I\,k\,a)}$$

$$\%2 := e^{(I\,k\,a - I\,\kappa\,a)}$$

```
> assign(sol2);
> A/C;
```

$$\frac{1}{4}(-e^{(I\,\kappa\,a + I\,k\,a)}\,\kappa^2 + e^{(I\,k\,a - I\,\kappa\,a)}\,\kappa^2 + e^{(I\,k\,a - I\,\kappa\,a)}\,k^2 - e^{(I\,\kappa\,a + I\,k\,a)}\,k^2 + 2\,k\,e^{(I\,k\,a - I\,\kappa\,a)}\,\kappa$$
$$+ 2\,k\,e^{(I\,\kappa\,a + I\,k\,a)}\,\kappa)/(k\,\kappa)$$

```
> A/B;
```

$$-\frac{-\%1\,\kappa^2+\%2\,\kappa^2+\%2\,k^2-\%1\,k^2+2\,k\,\%2\,\kappa+2\,k\,\%1\,\kappa}{\%2\,\kappa^2-\%2\,k^2+\%1\,k^2-\%1\,\kappa^2}$$

$$\%1 := e^{(I\,\kappa\,a+I\,k\,a)}$$

$$\%2 := e^{(I\,k\,a-I\,\kappa\,a)}$$

でほぼ求まっています．ここからの式の単純化は苦労します．以下では簡単に求めているように見えますが，ここまで来るのに大分試行錯誤しています．手でやるほうが早いですが，exp と trigonal 関数の間の対応や分母の有理化などを扱うときや，式が膨大な場合は Maple を補助的に使うと計算間違いが少なくすみます．もっとうまいやり方がある場合は教えて下さい．先ずは B/A で計算を進めていきます．

```
> BB:=simplify(convert(A/B,trig));
```

$$BB := -(\kappa^2\,\%2+I\,\kappa^2\,\%1-\kappa^2\,\%4+I\,\kappa^2\,\%3-k^2\,\%4+I\,k^2\,\%3+k^2\,\%2+I\,k^2\,\%1$$
$$-2\,k\,\kappa\,\%4+2\,I\,k\,\kappa\,\%3-2\,k\,\kappa\,\%2-2\,I\,k\,\kappa\,\%1)\Big/($$
$$-\kappa^2\,\%4+I\,\kappa^2\,\%3+k^2\,\%4-I\,k^2\,\%3-k^2\,\%2-I\,k^2\,\%1+\kappa^2\,\%2+I\,\kappa^2\,\%1)$$

$$\%1 := \sin(\kappa\,a+k\,a)$$

$$\%2 := \cos(\kappa\,a+k\,a)$$

$$\%3 := \sin(-k\,a+\kappa\,a)$$

$$\%4 := \cos(-k\,a+\kappa\,a)$$

振幅の二乗を求めるために，複素共役を取ってみましょう．

```
> conjugate(BB);
```

$$-\mathrm{conjugate}((\kappa^2\,\%2+I\,\kappa^2\,\%1-\kappa^2\,\%4+I\,\kappa^2\,\%3-k^2\,\%4+I\,k^2\,\%3+k^2\,\%2+I\,k^2\,\%1$$
$$-2\,k\,\kappa\,\%4+2\,I\,k\,\kappa\,\%3-2\,k\,\kappa\,\%2-2\,I\,k\,\kappa\,\%1)\Big/($$
$$-\kappa^2\,\%4+I\,\kappa^2\,\%3+k^2\,\%4-I\,k^2\,\%3-k^2\,\%2-I\,k^2\,\%1+\kappa^2\,\%2+I\,\kappa^2\,\%1))$$

$$\%1 := \sin(\kappa\,a+k\,a)$$

$$\%2 := \cos(\kappa\,a+k\,a)$$

$$\%3 := \sin(-k\,a+\kappa\,a)$$

$$\%4 := \cos(-k\,a+\kappa\,a)$$

うまくいきません．これは Maple が数の範囲を default で複素数まで考えているためです．そこで，k, κ, a などの変数を実数であると Maple にあらかじめ教えておきます．コマンドは

```
> assume(k,real):
> assume(kappa,real):
> assume(a,real):
```

です．以下の出力では assume がかかっている変数には〜(チルデ) がついています．simplify の二つ目の変数は side relations(副関係式) をあらわに指定しています．

```
> simplify(conjugate(BB)*BB,{cos(kappa*a)^2=1-sin(kappa*a)^2} );
```

$$\frac{4\,k^{\sim 2}\,\kappa^{\sim 2}+(\kappa^{\sim 4}-2\,k^{\sim 2}\,\kappa^{\sim 2}+k^{\sim 4})\sin(\kappa\tilde{}\,a\tilde{})^2}{\sin(\kappa\tilde{}\,a\tilde{})^2\,(\kappa^{\sim 4}-2\,k^{\sim 2}\,\kappa^{\sim 2}+k^{\sim 4})}$$

```
> collect(%,sin(kappa*a));
```

$$1+\frac{4\,k^{\sim 2}\,\kappa^{\sim 2}}{(\kappa^{\sim 4}-2\,k^{\sim 2}\,\kappa^{\sim 2}+k^{\sim 4})\sin(\kappa\tilde{}\,a\tilde{})^2}$$

次に A/C を求めていきます.

```
> CC:=convert(A/C,trig);
```

$$CC := \frac{1}{4}\,\frac{-\%1\,\kappa^{\sim 2}+\%2\,\kappa^{\sim 2}+\%2\,k^{\sim 2}-\%1\,k^{\sim 2}+2\,k\tilde{}\,\%2\,\kappa\tilde{}+2\,k\tilde{}\,\%1\,\kappa\tilde{}}{k\tilde{}\,\kappa\tilde{}}$$

$$\%1 := \cos(\kappa\tilde{}\,a\tilde{}+k\tilde{}\,a\tilde{})+I\sin(\kappa\tilde{}\,a\tilde{}+k\tilde{}\,a\tilde{})$$

$$\%2 := \cos(k\tilde{}\,a\tilde{}-\kappa\tilde{}\,a\tilde{})+I\sin(k\tilde{}\,a\tilde{}-\kappa\tilde{}\,a\tilde{})$$

```
> simplify(expand(conjugate(CC)*CC));
```

$$-\frac{1}{4}\,\frac{-k^{\sim 4}+k^{\sim 4}\cos(\kappa\tilde{}\,a\tilde{})^2-2\,k^{\sim 2}\,\kappa^{\sim 2}\cos(\kappa\tilde{}\,a\tilde{})^2-\kappa^{\sim 4}+\kappa^{\sim 4}\cos(\kappa\tilde{}\,a\tilde{})^2-2\,k^{\sim 2}\,\kappa^{\sim 2}}{k^{\sim 2}\,\kappa^{\sim 2}}$$

```
> simplify(%,{cos(kappa*a)^2=1-sin(kappa*a)^2});
> collect(%,sin(kappa*a));
```

$$\frac{1}{4}\,\frac{4\,k^{\sim 2}\,\kappa^{\sim 2}+(\kappa^{\sim 4}-2\,k^{\sim 2}\,\kappa^{\sim 2}+k^{\sim 4})\sin(\kappa\tilde{}\,a\tilde{})^2}{k^{\sim 2}\,\kappa^{\sim 2}}$$

$$\frac{1}{4}\,\frac{(\kappa^{\sim 4}-2\,k^{\sim 2}\,\kappa^{\sim 2}+k^{\sim 4})\sin(\kappa\tilde{}\,a\tilde{})^2}{k^{\sim 2}\,\kappa^{\sim 2}}+1$$

で大体もとまりました.

次に先ほど示した，透過率の plot を試みてみましょう．先ず求める関数を打ち込んでみます.

```
> CC3:=epsilon->(1+V0^2*sinh(alpha*a)^2/4/epsilon/(V0-epsilon))^(-1);
```

$$CC3 := \varepsilon \to \frac{1}{1+\dfrac{\frac{1}{4}\,V0^2\sinh(\alpha\,a)^2}{\varepsilon\,(V0-\varepsilon)}}$$

これだけでは変数の値が定まっていませんのプロットできません．条件 $mV_0a^2/\hbar^2 = 8$ から m を消去することを試みます.

```
> eq5:=m*V0*a^2/hbar^2=8;
```

$$eq5 := \frac{m\;V0\;a^{\sim 2}}{hbar^2} = 8$$

```
> alpha:=sqrt(2*m/hbar^2*(V0-epsilon));
```

$$\alpha := \sqrt{2}\,\sqrt{\frac{m\,(V0-\varepsilon)}{hbar^2}}$$

```
> solm:=solve({eq5},m);
```

$$solm := \{m = 8\,\frac{hbar^2}{V0\;a^{\sim 2}}\}$$

これを単純化するうまい方法が見つけられませんでした．仕方なく，二乗して V_0 を置き換えて，さらにルートを取りました.

```
> alpha:=sqrt(expand(subs(V0=1/iV0,(subs(solm,(alpha*a)^2)))));
```

$$\alpha := 4\sqrt{1 - iV0\,\varepsilon}$$

先程の関数の一部を取りだして書き換えます．ここで，$\alpha * a = \alpha$ と代えています．

```
> fun:=subs(V0=1/iV0,V0^2*sinh(alpha)^2/4/epsilon/(V0-epsilon));
```

$$fun := \frac{1}{4}\,\frac{\sinh(4\sqrt{1 - iV0\,\varepsilon})^2}{iV0^2\,\varepsilon\,(\frac{1}{iV0} - \varepsilon)}$$

このままではうまく単純化できません．分母と分子を別々に扱います．

```
> den_f:=denom(fun)/iV0;
```

$$den\_f := 4\,iV0\,\varepsilon\,(-1 + iV0\,\varepsilon)$$

```
> num_f:=numer(fun)/iV0;
```

$$num\_f := -\sinh(4\sqrt{1 - iV0\,\varepsilon})^2$$

iVo*epsilon を置き換えます．

```
> den_f1:=subs(iV0*epsilon=x,iV0^2*epsilon^2=x^2,expand(den_f));
```

$$den\_f1 := -4\,x + 4\,x^2$$

```
> num_f1:=subs(iV0*epsilon=x,num_f);
```

$$num\_f1 := -\sinh(4\sqrt{1 - x})^2$$

unapply 関数でユーザー関数とします．

```
> CC3:=unapply((1+num_f1/den_f1)^(-1),x);
```

$$CC3 := x \to \frac{1}{1 - \dfrac{\sinh(4\sqrt{1 - x})^2}{-4\,x + 4\,x^2}}$$

```
> plot(CC3(x),x=0..10,labels=["E/V0","|C/A|^2"]);
```

で先程の透過率の図が plot されます．

2.3 熱膨張係数の導出（複雑な関数の近似と積分）

（参考 キッテル著 固体物理学入門 宇 野良清他訳，丸善 1978 ）

基礎

熱膨張 (thermal expansion) は原子間ポテンシャルの二次以上の項によって現れます．平衡点からの原子の変位 x のポテンシャルエネルギーを

$$U(x) = cx^2 - gx^3 \tag{2.9}$$

と取ることができます．二次までの項では古典的な調和振動子を表し，熱膨張は現れません．x^3 の項は原子間相互作用の非対称性を表し，この項が熱膨張係数と直接かかわってきます．

有限温度での平均の変位は，ボルツマン分布関数を計算することで求まります．平均の位置 x は，熱力学的な確率で重みづけられ，

$$\langle x \rangle = \frac{\int_{-\infty}^{\infty} x \exp\left(-\beta U(x)\right) dx}{\int_{-\infty}^{\infty} \exp\left(-\beta U(x)\right) dx} \tag{2.10}$$

で計算できます．ここで $\beta \equiv 1/(k_{\mathrm{B}}T)$ です．この積分を実行すると

$$\langle x \rangle = \frac{3g}{4c^2} k_{\mathrm{B}} T \tag{2.11}$$

となります．

実践

最後の導出が問題です．ここでは先ず近似によって式を簡単化します．これはお決まりの Taylor 展開です．味噌はポテンシャルエネルギーの項で二次の項はそのまま残し，三次以上を展開することです．つまり三次以上の変化が $1/k_{\mathrm{B}}T$ よりも小さいと考えると展開が可能です．実際に Maple で展開してみると，

```
> restart;
> series(exp(-beta*(-g*x^3)),x,5);
```

$$1 + \beta\, g\, x^3 + \mathrm{O}(x^5)$$

です．これから関数を取りだします．

```
> temp:=convert(%,polynom);
```

$$temp := 1 + \beta\, g\, x^3$$

```
> f1:=unapply(exp(-beta*c*x^2)*temp,x);
```

$$f1 := x \rightarrow e^{(-\beta\, c\, x^2)} \left(1 + \beta\, g\, x^3\right)$$

$-\alpha$ から α まで積分して無限大まで持っていきます. (内側のコマンドから順に確認しながら打ち込んで下さい. 式をどのように変形していくかが分かります)

```
> limit(int(f1(x),x=-alpha..alpha),alpha=infinity);
```

$$\lim_{\alpha\to\infty} \frac{\sqrt{\pi}\,\mathrm{erf}(\sqrt{\beta\,c}\,\alpha)}{\sqrt{\beta\,c}}$$

うまく計算してくれません. これは βc が非負である必要があるためです. assume を用いて Maple に教えてやります.

```
> assume(beta>0);
> assume(c>0);
```

としておきます. 実際の積分を実行してみましょう. 先ず分母は,

```
> limit(int(f1(x),x=-alpha..alpha),alpha=infinity);
```

$$\frac{\sqrt{\pi}\,\sqrt{\beta\tilde{}\,c\tilde{}}}{\beta\tilde{}\,c\tilde{}}$$

次に分子は

```
> limit(normal(int(x*f1(x),x=-alpha..alpha)),alpha=infinity);
```

$$\frac{3}{4}\,\frac{\sqrt{\pi}\,\sqrt{\beta\tilde{}\,c\tilde{}}\,g}{\beta\tilde{}^{2}\,c\tilde{}^{3}}$$

です. これを取りだして (ここでは 2 回前を参照する%%を使っていますがはじめから何かの名前をつけておくほうが安全です), 簡単化します.

```
> den:=%%;
```

$$den := \frac{\sqrt{\pi}\,\sqrt{\beta\tilde{}\,c\tilde{}}}{\beta\tilde{}\,c\tilde{}}$$

```
> num:=%%;
```

$$num := \frac{3}{4}\,\frac{\sqrt{\pi}\,\sqrt{\beta\tilde{}\,c\tilde{}}\,g}{\beta\tilde{}^{2}\,c\tilde{}^{3}}$$

```
> simplify(num/den);
```

$$\frac{3}{4}\,\frac{g}{\beta\tilde{}\,c\tilde{}^{2}}$$

2.4 陽電子消滅寿命（連立微分方程式）

次の微分方程式の一般解を求めよ.

$$\left(\frac{\partial}{\partial t}n(t)\right)+\lambda n(t)=0 \tag{2.12}$$

さらに

$$\begin{aligned}\left(\frac{\partial}{\partial t}n_1(t)\right)+(\lambda_1+K_{12}n_1(t)) &= 0 \\ \left(\frac{\partial}{\partial t}n_2(t)\right)+\lambda_2 n_2(t) &= K_{12}n_1(t)\end{aligned} \tag{2.13}$$

の連立微分方程式を $n_1(0)=n_0, n_2(0)=0$ を初期条件として求めよ.

基礎

これは固体中に入った陽電子がとる状態の寿命を記述する微分方程式です. 陽電子は，あるひとつのサイト上で (2.12) 式のような微分方程式にしたがった寿命を示します. しかし，複数のサイトがあると，遷移確率を $K_{ij}(K_{ij}\neq K_{ji})$ とした場合，正確には

$$\frac{d}{dt}n_i(t)+\left(\lambda_i+\sum_{j\neq i}^{N}K_{ij}\right)n_i(t)=\sum_{j\neq i}^{N}K_{ji}n_j(t) \tag{2.14}$$

なる連立微分方程式で表す必要があります. しかし，これを解けば分かるように，2 つのサイトだけで陽電子が消滅すると仮定しても，表式は相当複雑になります. そこで，trapping model が提唱され，それに基づいて実験結果の解析がおこなわれています. このモデルでは，ある時刻 $t=0$ で全ての陽電子がひとつの状態に居り，そこで消滅するかまたは他のサイトへ遷移していくと考えます. 遷移した状態からさらに他の状態への遷移は小さいと考えると相当単純になります. ここで消滅サイトが 2 つだとすると (2.13) 式が得られます.（参考”Positron in Solids”,ed by P.Hautojarvi,Springer- Verlag,Berlin 1979 ）

実践

まずは単純な微分方程式の結果を見ておきます.

```
> restart;
> deq:={diff(n(t),t)+lambda*n(t)=0};
```

$$deq := \{(\tfrac{\partial}{\partial t}\,\mathrm{n}(t))+\lambda\,\mathrm{n}(t)=0\}$$

```
> dsolve(deq,n(t));
```

$$\mathrm{n}(t)=\_C1\,e^{(-\lambda\,t)}$$

次に課題の連立微分方程式をつくります．このとき，初期条件も入れておきます．

```
> deq:={diff(n[1](t),t)+(lambda[1]+K[12])*n[1](t)=0,
> diff(n[2](t),t)+lambda[2]*n[2](t)=K[12]*n[1](t),
> n[1](0)=n[0],n[2](0)=0};
```

$$deq := \{ (\tfrac{\partial}{\partial t} n_1(t)) + (\lambda_1 + K_{12})\, n_1(t) = 0,\ n_1(0) = n_0,\ n_2(0) = 0,\ (\tfrac{\partial}{\partial t} n_2(t)) + \lambda_2\, n_2(t) = K_{12}\, n_1(t) \}$$

実際に dsolve で解かせてみます．

```
> sol1:=dsolve(deq,{n[1](t),n[2](t)});
```

$$sol1 := \left\{ n_2(t) = \frac{K_{12}\, n_0\, e^{(-\lambda_2\, t + t\, \%1)} + \frac{e^{(-\lambda_2\, t)}\, K_{12}\, n_0\, \lambda_1}{\%1} + \frac{e^{(-\lambda_2\, t)}\, {K_{12}}^2\, n_0}{\%1} - \frac{e^{(-\lambda_2\, t)}\, K_{12}\, n_0\, \lambda_2}{\%1}}{\%1},\ n_1(t) = n_0\, e^{(-(\lambda_1 + K_{12})\, t)} \right\}$$

$$\%1 := -\lambda_1 - K_{12} + \lambda_2$$

$n_2(t)$ が一見ややこしそうです．assign して単純化しておきます．

```
> assign(sol1);
> simplify(n[2](t));
```

$$\frac{K_{12}\, n_0\, (e^{(-(\lambda_1 + K_{12})\, t)} - e^{(-\lambda_2\, t)})}{-\lambda_1 - K_{12} + \lambda_2}$$

2.5 Hamiltonian の解（線形代数）

基礎

異核 s 価電子ボンド 2 量体の原子軌道の線形結合から求めた永年方程式は

$$\begin{pmatrix} -\frac{1}{2}\Delta E-(E-\bar{E}) & h-(E-\bar{E})S \\ h-(E-\bar{E})S & +\frac{1}{2}\Delta E-(E-\bar{E}) \end{pmatrix}\begin{pmatrix} c_{\mathrm{A}} \\ c_{\mathrm{B}} \end{pmatrix}=0 \qquad (2.15)$$

である．ここで，$\Delta E = E_{\mathrm{A}} - E_{\mathrm{B}}$ は原子エネルギー準位のづれ，$\bar{E} = 1/2(E_{\mathrm{A}}+E_{\mathrm{B}})$，$h$ と S はそれぞれ軌道間のボンド積分と重なり積分を表す．この式を解き，S の 2 次以上の項を無視することによって結合状態 (bond) と反結合状態 (antibond) の固有値

$$E^{\pm}=\bar{E}+|h|S\mp\frac{1}{2}\sqrt{4h^2+(\Delta E)^2} \qquad (2.16)$$

を導け．（参考 分子・固体の結合と構造，ディヴィット・ペティフォー著，技報堂 1997 ）

実践

```
> diag1:=-1/2*DE-(E1-Eav);
> diag2:=1/2*DE-(E1-Eav);
> offdiag:=h-(E1-Eav)*S;
```

$$diag1 := -\frac{1}{2}\,DE - E1 + Eav$$

$$diag2 := \frac{1}{2}\,DE - E1 + Eav$$

$$offdiag := h - (E1 - Eav)\,S$$

```
> ham1:=<<diag1,offdiag>|<offdiag,diag2>>;
```

$$ham1 := \begin{bmatrix} -\frac{1}{2}\,DE - E1 + Eav & h - (E1 - Eav)\,S \\ h - (E1 - Eav)\,S & \frac{1}{2}\,DE - E1 + Eav \end{bmatrix}$$

```
> sol1:=Determinant(ham1);
```

$$sol1 := -\frac{1}{4}\,DE^2 + E1^2 - 2\,E1\,Eav + Eav^2 - h^2 + 2\,h\,S\,E1 - 2\,h\,S\,Eav - S^2\,E1^2 + 2\,S^2\,E1\,Eav - S^2\,Eav^2$$

```
> solve(sol1=0,E1);
```

$$\frac{1}{2}\,\frac{-8\,Eav + 8\,S^2\,Eav + 8\,h\,S + 4\sqrt{DE^2 + 4\,h^2 - S^2\,DE^2}}{-4 + 4\,S^2},$$
$$\frac{1}{2}\,\frac{-8\,Eav + 8\,S^2\,Eav + 8\,h\,S - 4\sqrt{DE^2 + 4\,h^2 - S^2\,DE^2}}{-4 + 4\,S^2}$$

```
> Ebond:=%[1];
```

$$Ebond := \frac{1}{2}\,\frac{-8\,Eav + 8\,S^2\,Eav + 8\,h\,S + 4\,\sqrt{DE^2 + 4\,h^2 - S^2\,DE^2}}{-4 + 4\,S^2}$$

```
> series(Ebond,S,2);
```

$$(Eav - \frac{1}{2}\,\sqrt{DE^2 + 4\,h^2}) - h\,S + \mathrm{O}(S^2)$$

2.6 オイラー角（線形代数）

中心点を固定した，図のような軸の周りの連続する回転を考え，２つの座標系の方向余弦をオイラー角で表せ.（参考：ベクトル・テンソルと行列，ジョージ・アルフケン著，講談社 1977 ）



図 2.1:

```
> with(LinearAlgebra):
> RZ1:=Matrix(3,3,[[cos(alpha),sin(alpha),0],[-sin(alpha),cos(alpha),0]
> ,[0,0,1]]);
```

$$RZ1 := \begin{bmatrix} \cos(\alpha) & \sin(\alpha) & 0 \\ -\sin(\alpha) & \cos(\alpha) & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

```
> RY:=Matrix(3,3,[[cos(beta),0,-sin(beta)],[0,1,0],[sin(beta),0,cos(bet
> a)]]);
```

$$RY := \begin{bmatrix} \cos(\beta) & 0 & -\sin(\beta) \\ 0 & 1 & 0 \\ \sin(\beta) & 0 & \cos(\beta) \end{bmatrix}$$

```
> RZ2:=Matrix(3,3,[[cos(gamma),sin(gamma),0],[-sin(gamma),cos(gamma),0]
> ,[0,0,1]]);
```

$$RZ2 := \begin{bmatrix} \cos(\gamma) & \sin(\gamma) & 0 \\ -\sin(\gamma) & \cos(\gamma) & 0 \\ 0 & 0 & 1 \end{bmatrix}$$

```
> Trans1:=RZ2.RY.RZ1;
```

$$Trans1 :=$$
$$[\cos(\alpha)\cos(\beta)\cos(\gamma) - \sin(\alpha)\sin(\gamma)\,,\ \sin(\alpha)\cos(\beta)\cos(\gamma) + \cos(\alpha)\sin(\gamma)\,,$$
$$-\sin(\beta)\cos(\gamma)]$$
$$[-\cos(\alpha)\cos(\beta)\sin(\gamma) - \sin(\alpha)\cos(\gamma)\,,\ -\sin(\alpha)\cos(\beta)\sin(\gamma) + \cos(\alpha)\cos(\gamma)\,,$$
$$\sin(\beta)\sin(\gamma)]$$
$$[\cos(\alpha)\sin(\beta)\,,\ \sin(\alpha)\sin(\beta)\,,\ \cos(\beta)]$$

2.7 組成自由エネルギー曲線（連立方程式の数値解）



上図は 2 成分系 1300K における，正則溶体近似に基づいた液相と固相の自由エネルギー組成曲線を表している．この曲線を表す関数を求めよ．また，ある温度での 2 相平衡の組成を求めよ.（参考：材料組織学，杉本孝一，長村光造，山根壽己，牧正志，菊池潮美，落合庄治郎，村上陽太郎著，朝倉書店 1991 ）

基礎

材料科学でよく出てくる状態図の求め方の基礎です．馴染みのない方がほとんどだと思いますので，詳しく解説しながら数値的に連立方程式を解く方法を見ていきます.

先ず状態図の求め方を逆順で書くと，

1. 状態図を求めることは二相平衡の条件を求めることである.
2. 固体と液体の化学ポテンシャルが等しい組性を求める.
3. 化学ポテンシャルは自由エネルギー関数の微分で求まる.
4. 正則溶体近似に基づいた自由エネルギー関数を求める.
5. 自由エネルギー関数に必要な値を仮定する.

となります．２．と３．とはまとめて共通接線を求めることと等価です.
まず４．の正則溶体近似に基づいた自由エネルギー関数は

```
> Gs:=(xb,temp)->(1-xb)*muas+xb*mubs+dGms(xb,temp);
Gl:=(xb,temp)->(1-xb)*mual+xb*mubl+dGml(xb,temp);
```

$$Gs := (xb,\ temp) \to (1 - xb)\ muas + xb\ mubs + \mathrm{dGms}(xb,\ temp)$$

$$Gl := (xb,\ temp) \to (1 - xb)\ mual + xb\ mubl + \mathrm{dGml}(xb,\ temp)$$

です．この中で dGms,dGml で表した混合の自由エネルギーは正則溶体では理想溶体の混合のエントロピーと相互作用パラメーター (omega) を用いて，

```
> dGms:=(xb,temp)->Omegas*(1-xb)*xb+RR*temp*(xb*ln(xb)+(1-xb)*ln(1-xb))
;
dGml:=(xb,temp)->Omegal*(1-xb)*xb+RR*temp*(xb*ln(xb)+(1-xb)*ln(1-xb));
```

$$dGms := (xb,\ temp) \to Omegas\,(1 - xb)\,xb + RR\,temp\,(xb\ln(xb) + (1 - xb)\ln(1 - xb))$$

$$dGml := (xb,\ temp) \to Omegal\,(1 - xb)\,xb + RR\,temp\,(xb\ln(xb) + (1 - xb)\ln(1 - xb))$$

となります．これらの式中で使われている定数と変数を以下のようにとります．

```
> RR:=8.314:
Omegas:=16.7*1000:  muas:=0:  mubs:=0:
Omegal:=0:
dHa:=8.37*1000:  dHb:=12.55*1000:
dSa:=dHa/1000:  dSb:=dHb/1500:
```

mual と mubl は融点において液相と固相の化学ポテンシャルの差が消えることからその温度依存を仮定します．

```
> mual:=temp->dHa-temp*dSa;
mubl:=temp->dHb-temp*dSb;
```

$$mual := temp \to dHa - temp\ dSa$$

$$mubl := temp \to dHb - temp\ dSb$$

これに伴って先程定義した液相の自由エネルギー関数の mu も温度に依存した関数として定義し直しておく必要が出てきます．

```
> Gl:=(xb,temp)->(1-xb)*mual(temp)+xb*mubl(temp)+dGml(xb,temp);
```

$$Gl := (xb,\ temp) \to (1 - xb)\,\mathrm{mual}(temp) + xb\,\mathrm{mubl}(temp) + \mathrm{dGml}(xb,\ temp)$$

これらの液相・固相の自由エネルギー関数がある一定温度でちゃんと再現されているか確かめておきましょう．

```
> templ:=1300;
plot({Gl(x,templ),Gs(x,templ)},x=0..1);
```

$$templ := 1300$$




```
> templ:=800;
plot({Gl(x,templ),Gs(x,templ)},x=0..1);
```

$$templ := 800$$



これで作業項目の４．５．が終わりました．あとは共通接線を求めればいいだけです．関数の微分が diff によって求められることを思い出して，

```
> diff(Gl(y,temp),y);
```

$$4180.00 + .003333333\, temp + 8.314\, temp\, (\ln(y) - \ln(1-y))$$

まず傾きの差を

```
> F1:=(x,y,temp)->subs(x1=x,y1=y,diff(Gs(x1,temp),x1)-diff(Gl(y1,temp),
y1));
```

$$F1 := (x,\, y,\, temp) \rightarrow \mathrm{subs}(x1 = x,\, y1 = y,\, (\tfrac{\partial}{\partial x1}\,\mathrm{Gs}(x1,\, temp)) - (\tfrac{\partial}{\partial y1}\,\mathrm{Gl}(y1,\, temp)))$$

次に切片の差は

```
> F2:=(x,y,temp)->subs(x1=x,y1=y,Gs(x1,temp)-diff(Gs(x1,temp),x1)*x1-(G
l(y1,temp)-diff(Gl(y1,temp),y1)*y1));
```

$$F2 := (x,\, y,\, temp) \rightarrow \mathrm{subs}(x1 = x,\, y1 = y,\, \mathrm{Gs}(x1,\, temp) - (\tfrac{\partial}{\partial x1}\,\mathrm{Gs}(x1,\, temp))\, x1 - \mathrm{Gl}(y1,\, temp) + (\tfrac{\partial}{\partial y1}\,\mathrm{Gl}(y1,\, temp))\, y1)$$

これを fsolve で連立方程式として，x ，y について数値的に求めてみましょう.

```
> fsolve({F1(x,y,1300)=0,F2(x,y,1300)=0},{x,y},{x=0.5..0.99,y=0.5.
.0.99});
```

$$\{x = .9685370576,\, y = .8308884633\}$$

となり，解が得られます．これを関数と定義します.

```
> FindXY:=(temp)->fsolve({F1(x,y,temp)=0,F2(x,y,temp)=0},{x,y},{x=
0.5..0.99,y=0.5..0.99});
```

$$FindXY := temp \rightarrow \mathrm{fsolve}(\{\mathrm{F1}(x,\, y,\, temp) = 0,\, \mathrm{F2}(x,\, y,\, temp) = 0\},\, \{x,\, y\},\, \{y = .5...99,\, x = .5...99\})$$

```
> FindXY(1300);
```

$$\{x = .9685370576,\, y = .8308884633\}$$

これを温度を順繰りに変えて求めます.

```
> for temp from 1400 by -100 to 900
do
print(temp);
FindXY(temp);
od;
```

$$1400$$
$$\{x = .9843946402,\, y = .9164387422\}$$
$$1300$$
$$\{x = .9685370576,\, y = .8308884633\}$$
$$1200$$
$$\{x = .9524858191,\, y = .7434269425\}$$
$$1100$$
$$\{x = .9363612791,\, y = .6542285850\}$$
$$1000$$
$$\{x = .9204118951,\, y = .5636183410\}$$
$$900$$

$$\begin{aligned}&\mathrm{fsolve}(\{-33400.0\,x+12517.00000+7482.600\ln(x)-7482.600\ln(1-x)\\&-7482.600\ln(y)+7482.600\ln(1-y)=0,16700.0\,(1-x)\,x+7482.600\,x\ln(x)\\&+7482.600\,(1-x)\ln(1-x)\\&-(-33400.0\,x+16700.000+7482.600\ln(x)-7482.600\ln(1-x))\,x\\&-837.000000-4183.000000\,y-7482.600\,y\ln(y)-7482.600\,(1-y)\ln(1-y)\\&+(4183.00000+7482.600\ln(y)-7482.600\ln(1-y))\,y=0\},\{x,\,y\},\\&\{y=.5...99,\,x=.5...99\})\end{aligned}$$

ちゃんと求まっているでしょうか？ このような作業も立派なプログラミングの基礎です．あとはこれを関数として全て定義してしまえば終わりです．しかし，最後のところでエラーが出ています．これがなぜ出たか考えてみてください．広い温度範囲で求めたり，純 A 組性に近いところでの固相ー液相平衡などに対応する必要があります．違ったアルゴリズムを使うか，条件判断を加えてより汎用性の高いプログラムを組むのが本来のプログラミングの難しいところです（本書の範囲を越えますのでここでは扱いません）．

2.8 減衰振動のフーリエ変換（高速フーリエ変換)

ファイル名 DATA102 は，鋼の棒をわずかに弾性変形して自由振動させたときの波形である．この振動にはどのような周波数成分が含まれているか，複素フーリエ変換により調べよ．固体を振動させると，振幅は大なり小なり減衰する．これは，固体中のさまざまな欠陥が弾性エネルギーを吸収して熱エネルギーに変えてしまうことに起因する．

```
> restart;
> T:=readdata(‘DATA102‘,1):
> with(plots):
> listplot(T);
```




```
> with(linalg):
> n:=vectdim(T);
> Rdata:=convert(T,array):
> Idata:=array(sparse,1..n);
```

```
Warning, new definition for norm

Warning, new definition for trace
```

$$n := 4096$$

$$Idata := \mathrm{array}(sparse,\ 1..4096,\ [])$$

```
> FFT(12,Rdata,Idata);
```

$$4096$$

```
> with(plots):
> datapoint:=[seq([i,sqrt(Rdata[i]^2+Idata[i]^2)],i=1..256)]:
> logplot(datapoint);
```




```
> datapoint:=[seq([i,Idata[i]/Rdata[i]],i=2..1024)]:
> plot(datapoint);
```



付 録A　コマンドのまとめ

本文で引用したコマンドをメモ代わりに載せておきます．徐々に充実させます．有益なヘルプは次の通りです．

? inifcns - 起動時から認識されている関数

? index[package] - 関連する関数を集めたパッケージです．微分方程式 (DETools), 線形代数 (linalg)，プロット関係の関数 (plots) 等があります．

? index[function] - Maple の標準関数．

A.1　基本操作のまとめ

```
> 1+1;
> #(return) 入力，改行は (shigt+return)
```

$$2$$

```
> plot(sin(x),x=-Pi..Pi):#プロット
> ?plot:#ヘルプの例
> mass:=10;#代入
> force:=-mass*accel;
```

$$mass := 10$$

$$force := -10\, accel$$

```
> mass:='mass';#代入の取り消し
```

$$mass := mass$$

```
> restart;#すべての代入のリセット
> force:=-mass*accel;
> subs(mass=10,accel=14,force);#一時的な代入
```

$$force := -mass\, accel$$

$$-140$$

```
> evalf(log[2](5),20);#数値へ変換 (20 桁)
```

$$2.3219280948873623479$$

```
> eq1:=exp(-x)*cos(10*x);
```

$$eq1 := e^{(-x)} \cos(10\, x)$$

```
> f1:=unapply(eq1,x);#関数の定義
```

$$f1 := x \to e^{(-x)} \cos(10\,x)$$

```
> plot(f1(z),z=0..infinity);
```




```
> with(plots):
> f2:=(x,t)->exp(-x)*cos(t*x);#関数の定義
> animate(f2(z,t),z=0..infinity,t=-10..10);#アニメ，出力画面
```
を選んでメニューの矢印を操作

$$f2 := (x,\, t) \to e^{(-x)} \cos(t\,x)$$




```
> eqset:={x+y=1,y=x^2+1}:
> solset:=solve(eqset,{x,y});#方程式の解 (数値解は fsolve)
> assign(solset[1]);#値の代入
```

$$solset := \{y = 1,\, x = 0\},\, \{y = 2,\, x = -1\}$$

```
> sum(i,i=1..10);#和
> add(i,i=1..10);#数値の和
```

$$55$$

$$55$$

```
> restart;
> limit(sin(x)/x,x=0);#極限
```

$$1$$

```
> diff(x^2*y,x);#微分
```

$$2\,x\,y$$

```
> eq:=diff(y(x),x)+y(x)=0;
> dsolve({eq,y(0)=a},y(x));#微分方程式
```

$$eq := (\tfrac{\partial}{\partial x}\,\mathrm{y}(x)) + \mathrm{y}(x) = 0$$

$$\mathrm{y}(x) = a\,e^{(-x)}$$

```
> int(x^2/sqrt(1-x^2),x=0..Pi);#定積分
```

$$-\frac{1}{2}\,I\,\sqrt{-1+\pi^2}\,\pi + \frac{1}{2}\,I\ln(\pi - \sqrt{-1+\pi^2}) + \frac{1}{4}\,\pi$$

```
> series(sin(x),x=Pi/2,4);#級数
```

$$1 - \frac{1}{2}\,(x - \frac{1}{2}\,\pi)^2 + \mathrm{O}((x - \frac{1}{2}\,\pi)^4)$$

```
> convert(%,polynom);#級数展開からの式の取りだし
```

$$1 - \frac{1}{2}\,(x - \frac{1}{2}\,\pi)^2$$

```
> list1:=[seq(i^2,i=1..4)];#リスト
```

$$list1 := [1,\ 4,\ 9,\ 16]$$

```
> nops(list1);
```

$$4$$

```
> list2:=[op(list1),17];#リストへの追加
```

$$list2 := [1,\ 4,\ 9,\ 16,\ 17]$$

```
> map(sqrt,list2);#リストへ作用
```

$$[1,\ 2,\ 3,\ 4,\ \sqrt{17}]$$

A.2 関数パッケージ

? inifcns - 起動時から認識されている関数

? integer - 整数関数

? polynom - 多項式の取り扱いに関する関数

? ratpoly - 有理関数

? vector - ベクトル関数

? matrix - 行列に関する関数

? numtheory - 整数論

? combinat - 組み合わせ数学

? stats - 統計学

? orthopoly - 直交多項式

A.3 式の変形

- expand - Expand an Expression （展開）
- simplify - Apply Simplification Rules to an Expression （簡単化，これですべてうまく行くといいのですが...）
- collect - Collect Coefficients of Like Powers（項の次数で式をまとめる）
- combine - combine terms into a single term （規則にしたがって式をまとめる）
- coeff - extract a coefficient of a polynomial（多項式の係数の取り出し）
- sort - sort a list of values or a polynomial （式や値のソート）
- factor - Factor a Multivariate Polynomial （一つ以上の変数を持つ多項式の因数分解）
- normal - normalize a rational expression （約分，通分）
- convert - convert an expression to a different form （関数を違う関数へ変換）
- gcd - greatest common divisor of polynomials （多項式の最大公約数）
- lcm - least common multiple of polynomials （多項式の最小公倍数）
- numer - numerator of an expression （分母）
- denom - denominator of an expression （分子）
- radsimp - simplification of an expression containing radicals （分母の有理化)
- assume - The Assume Facility （変数の範囲や関係を規定）

A.4 線形代数 (linalg)

linalg の使い方をまとめておきます．LinearAlgebra との使い分けは付録Dを参照ください．

matrix や vector の生成は，

```
> with(linalg):
> ma:=matrix(2,2,[1,2,3,4]);
```

$$ma := \begin{bmatrix} 1 & 2 \\ 3 & 4 \end{bmatrix}$$

```
> vec:=vector([x,y]);
```

$$vec := [x, y]$$

等です．ここで Vector は縦（列）ベクトル (column vector) を生成することに注意ください．用意されている演算の主なものと以下に掲げておきます．

- matadd - matrix or vector addition（和）(matadd(A,B,c1,c2):c1*A+c2*B)
- dotprod - vector dot (scalar)product （内積）
- crossprod - vector cross product （外積）
- inverse - compute the inverse of a matrix （逆行列）
- det - determinant of a matrix （行列式）
- trace - the trace of a matrix （対角和）
- adjoint - compute the adjoint of a matrix （随伴）
- transpose - compute the transpose of a matrix （転置行列）
- eigenvals - compute the eigenvalues of a matrix （固有値）
- eigenvects - find the eigenvectors of a matrix （固有ベクトル）
- angle - compute the angle between vectors （角度）
- rowdim - determine the row dimension of a matrix （行の数）
- coldim - determine the column dimension of a matrix （列の数）
- vectdim - determine the dimension of a vector （ベクトルの要素数）

この他にも curl,diverge,grad,jacobian などの微分演算子や，行列やベクトルの行や列を操作するオペレーションが用意されています．

付 録B　入出力に関する補足

Unix での使用の際にはフィルターとして Maple が使えると便利です．また，Maple にはプログラム言語への翻訳や L^AT_EX への出力機能が備わっています．これらを簡単に紹介しておきます．

B.1　データの入出力 (フィルターとしての使用)

unix の標準的な使い方として，maple -f action.txt ¡ input.data ¿output.data というのが用意されていればいいのですが，どうやら現 version ではなさそうです．そこでそれに近い動作をさせるスクリプトを示し，あわせて mint の使い方を御紹介します．

以下の例は ‘DATA101‘中のデータを読み込んで，‘out.data‘に書き込むスクリプトです．editor で以下のような入力スクリプトを用意します．

```
[root@asura bin]# cat test.txt
T:=readdata('/home/bob/DATA101',float,1):
writedata('/home/bob/out.data',T);
quit:
```

非常に単純なスクリプトで書けます．これが動くかを確認するために，mint でスクリプトのチェックをしてみました．上記のスクリプトにはバグが潜んでいます．

```
[root@asura bin]# mint<test.txt
    |\^/|      Maple 6 Diagnostic Program
._|\|   |/|_.  Copyright (c) 1981-1998 by Waterloo Maple Inc. All rights
 \  MINT   /   reserved. Maple and Maple V are registered trademarks of
 <____ ____>   Waterloo Maple Inc.
      |
on line     1: T:=readdata('/home/bob/DATA101',float,1):
                           ^ syntax error
on line     2: writedata('/home/bob/out.data',T);
                         ^ syntax error
```

mint は二つの syntax error を出しています．syntax error はバックコートで

囲むべき文字列をシングルコートで囲んだ事による error です．これを訂正して，走らせてみると，

```
[root@asura bin]# maple < test2.txt
    |\^/|     Maple V Release 5 (Kyoto University)
._|\|   |/|_. Copyright (c) 1981-1997 by Waterloo Maple Inc. All rights
 \  MAPLE  /  reserved. Maple and Maple V are registered trademarks of
 <____ ____>  Waterloo Maple Inc.
      |       Type ? for help.
> T:=readdata(‘/home/bob/DATA101‘,float,1):
> writedata(‘/home/bob/out.data‘,T);
> quit:
bytes used=432084, alloc=393144, time=0.07

[root@asura bin]# wc /home/bob/out.data
    256     256    3577 /home/bob/out.data
```

めでたく出力されています．

B.2 C, LATEX への出力

Maple で作った式や関数などは，Fortran や C, L^AT_EX 等で使える format に変換することが可能です．以下にその例をお見せしておきます．

```
> series(exp(-beta*(-g*x^3)),x,5);
> temp:=convert(%,polynom);
> f1:=unapply(exp(-beta*c*x^2)*temp,x);
> eqn1:=simplify(limit(int(f1(x),x=-alpha..alpha),alpha=infinity));
```

$$1 + \beta\, g\, x^3 + \mathrm{O}(x^5)$$

$$temp := 1 + \beta\, g\, x^3$$

$$f1 := x \to e^{(-\beta\, c\, x^2)}\,(1 + \beta\, g\, x^3)$$

$$eqn1 := \lim_{\alpha\to\infty} \frac{\sqrt{\pi}\,\mathrm{erf}(\sqrt{\beta\, c}\,\alpha)}{\sqrt{\beta\, c}}$$

```
> with(codegen,C):
> C(f1(x),optimized);
      t2 = x*x;
      t4 = exp(-beta*c*t2);
      t9 = t4*(1.0+beta*g*t2*x);
> latex(eqn1);
\lim _{\alpha\rightarrow \infty }{\frac {\sqrt {\pi }{\it erf}(\sqrt {
\beta\,c}\alpha)}{\sqrt {\beta\,c}}}
```

付 録C　plotのまとめ

特殊な plot や，図の作り方を例示します.

- 少し手の込んだ概略図の作成
- データの等高線表示
- 分子の表示

C.1　少し手の込んだ概略図の作成

```
> restart;
> with(plots):
> with(plottools):
```

```
Warning, the name changecoords has been redefined
```

```
> V:=x->piecewise(x>=0 and x<=1,1);
> p1:=plot('V(x)',x=-1..2,y=0..1.5,thickness=4):
> display({p1},labels=[distance,energy],
> labeldirections=[HORIZONTAL,VERTICAL],
> labelfont=[TIMES,ROMAN,22],
> title=`Potential function for the tunnel effect`,
> titlefont=[TIMES,ROMAN,22],tickmarks=[3,3]);
```

$$V := x \rightarrow \text{piecewise}(0 \leq x \textbf{ and } x \leq 1,\ 1)$$



```
> arrow1:=PLOT(arrow( [-1,0.8], [-0.2,0.8], 0.005,0.1,1/3,color=black)
> ):
> arrow2:=PLOT(arrow( [-0.2,0.4], [-1,0.4], 0.005,0.1,1/3,color=black)
> ):
> arrow3:=PLOT(arrow( [1.2,0.6], [2,0.6], 0.005,0.1,1/3,color=black)
):
> arrowx:=PLOT(arrow( [-1,0], [2,0], 0.0005,0.1,1/20, color=black)
):
> arrowy:=PLOT(arrow( [0,0], [0,1.5], 0.001,0.1,1/10, color=black)
):
> AA:=textplot([-0.6,0.9,"A"],font=[HELVETICA,OBLIQUE,22]):
> BB:=textplot([-0.6,0.5,"B"],font=[HELVETICA,OBLIQUE,22]):
> CC:=textplot([1.6,0.7,"C"],font=[HELVETICA,OBLIQUE,22]):
> V0:=textplot([-0.25,1.0,"V0"],font=[HELVETICA,OBLIQUE,22]):
> zero:=textplot([0,-0.1,"0"],font=[HELVETICA,OBLIQUE,22]):
> aa:=textplot([1,-0.1,"a"],font=[HELVETICA,OBLIQUE,22]):
> display({p1,arrow1,arrow2,arrow3,AA,BB,CC,V0,zero,aa,arrowx,arrowy},
> axes=NONE,view=[-1..2,-0.1..1.5],title=‘Potential function
for the
> tunnel effect‘,
> titlefont=[TIMES,ROMAN,22]);
```



C.2 データの等高線表示

```
> restart;
> #L:=readdata("E-TS_lam12.txt",4):
> p2:=[seq([seq(L[i*6+j+1][4],j=0..5)],i=0..6)];
```

$p2 := [[-6.548925792, -6.556969238, -6.569859881, -6.571725175, -6.552130841, -6.507429625], [-6.549154934, -6.558561941, -6.576399575, -6.585797044, -6.575353151, -6.539476937], [-6.549040863, -6.559647139, -6.581471285, -6.597624243, -6.595363553, -6.568589566], [-6.548432274, -6.559996663, -6.585129484, -6.606877007, -6.612371244, -6.594356013], [-6.546440626, -6.558680986, -6.585765397, -6.611685294, -6.623831734, -6.614838616], [-6.541757939, -6.553566103, -6.581530372, -6.610083243, -6.628005998, -6.628118908], [-6.532058732, -6.543141924, -6.570227975, -6.600338521, -6.623502987, -6.632769581]]$

```
> with(plots):
> listplot3d(p2,axes=boxed,orientation=[-126,44]);
```




```
> listcontplot3d(p2,levels=12,axes=boxed,orientation=[-126,44],filled=t
> rue);
```




```
> listcontplot(p2,levels=12,filled=true);
```



C.3 分子の表示

```
> unit:=[[0,0,0],[1/2,1/2,0],[1/2,0,1/2],[0,1/2,1/2]];
> with(linalg):n:=1;
> period:=[seq(seq(seq([i,j,k],i=-n..n),j=-n..n),k=-n..n)]:
> period_n:=nops(period);
> L1:=[seq(seq(matadd(unit[i],period[j],1,1),i=1..4),j=1..period_n)]:
> L_max:=nops(L1);
> L2:=[[0,0,0]];
> for j from 1 to L_max do
> poi:=L1[j]-[0,0,0];
> dist:=evalf(dotprod(poi,poi));
> if (dist < 0.8^2) and (dist > 0.1) then
> L2:=[op(L2),convert(L1[j],list)];
> fi;
> od:
> nops(L2);
> L:=convert(L2,listlist);
```

$$unit := [[0, 0, 0], [\frac{1}{2}, \frac{1}{2}, 0], [\frac{1}{2}, 0, \frac{1}{2}], [0, \frac{1}{2}, \frac{1}{2}]]$$

$$n := 1$$

$$period\_n := 27$$

$$L\_max := 108$$

$$L2 := [[0, 0, 0]]$$

$$13$$

$$L := [[0, 0, 0], [0, \frac{-1}{2}, \frac{-1}{2}], [\frac{-1}{2}, 0, \frac{-1}{2}], [\frac{1}{2}, 0, \frac{-1}{2}], [0, \frac{1}{2}, \frac{-1}{2}], [\frac{-1}{2}, \frac{-1}{2}, 0], [\frac{1}{2}, \frac{-1}{2}, 0], [0, \frac{-1}{2}, \frac{1}{2}], [\frac{-1}{2}, \frac{1}{2}, 0], [\frac{-1}{2}, 0, \frac{1}{2}], [\frac{1}{2}, \frac{1}{2}, 0], [\frac{1}{2}, 0, \frac{1}{2}], [0, \frac{1}{2}, \frac{1}{2}]]$$

```
> with(plottools):
> for i from 1 to 12 do
> c||i := sphere(L[i+1], .1,color=red):
> od:
> c0 := sphere([0,0,0], .1,color=blue):
> display([c0,c1,c2,c3,c4,c5,c6,c7,c8,c9,c10,c11,c12],
> scaling=constrained, style=patchnogrid,
> axes=boxed,orientation=[-126,44]);
```




```
> D1:=polyhedraplot(L,polyscale=.1,polytype=icosahedron,color=red):
> D2:=polyhedraplot([[0,0,0]],polyscale=.1,polytype=icosahedron,color=bl
> ue):
> display(D1,D2,axes=normal,scaling=constrained,
> style=patch,shading=NONE,orientation=[-126,44],labels=[x,y,z]);
```




```
> ?vrml;
```

付 録D　Maple6での改良点について

Maple の中身は V から 6 へのバージョンアップ (2000 年春) によってだいぶ変わったようです．以下にヘルプで得られる情報を元にまとめておきます．

- 旧来の関数などで変わった箇所
- LinearAlgebra
- module(未)
- 外部関数の呼び出し (未)

D.1　旧来の関数などで変わった箇所

?compatibility でくわしい情報が見れます．以下は本書に関係のある変更箇所です．

- C, fortran は codegen[C], codegen[fortran] などに変更．
- 連結作用素 “.” が “||” になった．
- readlib は必要なくなった．必要な関数は自動的に呼び出される．

D.2　LinearAlgebra

Maple の線形代数計算にはパッケージとして linalg と LinearAlgebra(LA) とがあります．従来のバージョンでは linalg のみでしたが，NAG との提携を期にあらたにデータ構造から関数操作まで作り直したようです．どちらを選ぶかは

linalg 抽象的な線形代数の計算向き．

LA 種々の形態の Matrix が生成しやすく，直観的な代数計算ができ，大きな行列の数値計算に向いている．

が基準となります．代数計算が直観的であることから今後 LA が標準となるようです．本書でも LA パッケージを使うよう書き換えています．

参考図書

Maple V に関する解説書が最近盛んに出版されています．一部を紹介します．

”MapleV による数式処理入門 ” 阿部寛著（講談社，1997)
間違いなく現在日本語で読める絶好の入門書．著者が長年 (たぶん？)，専門で使い込いこまれた経験が凝縮されています．特に微分方程式の具体的な解き方は充実しています．リリース 5 でもほとんど書き換える必要なく使えます．

”はじめての MapleV リリース 4 ” K.M. ヒール，M.L. ハンセン，K.M. リカード著，笠島友美訳（シュープリンガー・フェアラーク東京，1997)
リリース 4 の”Maple V Learning Guide”の訳．リリース 3 やリリース 5 の日本語版がいかにもマニュアルの翻訳という雰囲気で読みにくいのに対して，リリース 4 版は数学を理解している人がコマンドを確認しながら翻訳されたらしく，とても理解しやすくなっています．リリース 5 でも使えます．

”MapleV リリース 5 ラーニングガイド ” K.M. ヒール，M.L. ハンセン，K.M. リカード著，示野信一他訳（シュープリンガー・フェアラーク東京，1998)
新しいリリース 5 の”Maple V Learning Guide”の訳．

“ MapleV リリース 5 プログラミングガイド ” M.B. モナガン他著（シュープリンガー・フェアラーク東京，1999)
プログラミングや Maple の内部構造などをより詳細に知りたいときには購入する必要があります．内容は一般的な help では得られないような Maple の概念を統一的に説明しています．ただ，記述が抽象的（数学的?）なため初心者には何を言っているのか全く理解できません．Maple V を系統的に理解し，効率良く使うには是非とも必要です．

”Maple V と利用の実際-数式処理と CG- ” 小国　力著（サイエンス社 1997)
多くの解説書 (MATLAB,Mathematice) を出している著者による Maple V 解説書．広い分野を網羅しており，経験のある研究者が具体的な問題を解くときの足掛かりとなる．

”Maple V で見る数学ワールド ” 示野信一著 (シュプリンガーフェアラーク社 1999)
Maple を使って広い分野の数学を視覚化しようという意欲的な著書. 内容は大学初学年制でも理解できるが, 使われている技巧は高度.

索 引